POP-2 PAPERS

Introduction and Reference Manual R. M. Burstall, J. S. Collins and R. J. Popplestone

POP-2 PAPERS

POP-2 PAPERS

INTRODUCTION TO POP-2

R. M. BURSTALL
AND
J. S. COLLINS

AND

POP-2
REFERENCE MANUAL

R. M. BURSTALL
AND
R. J. POPPLESTONE

OLIVER & BOYD Edinburgh and London

OLIVER AND BOYD LTD Tweeddale Court, Edinburgh 1 39a Welbeck Street, London W.1

This document and all software referred to therein is copyright of the Round Table, Department of Machine Intelligence and Perception, University of Edinburgh

First published 1968 © 1968 The Round Table

Printed in Great Britain by T. and A. Constable Ltd Edinburgh

AN INTRODUCTION TO THE POP-2 PROGRAMMING LANGUAGE

BY

R. M. BURSTALL

AND

J. S. COLLINS

DEPARTMENT OF MACHINE INTELLIGENCE AND PERCEPTION UNIVERSITY OF EDINBURGH

CONTENTS

1	INTRODUCTION	3
2	ARITHMETIC AND THE STACK	4
3	SIMPLE VARIABLES	5
4	FUNCTIONS	5
5	CONDITIONALS, LOOPS AND LABELS	7
6	ARRAYS	9
7	LISTS AND NON-NUMERICAL VALUES	11
8	RECORDS	13
9	INPUT AND OUTPUT FACILITIES	14
10	ADDITIONAL EXERCISES	16
11	ADVANCED FEATURES	18

1. INTRODUCTION

POP-2 is a programming language based on R. J. Popplestone's POP-1 (1968) and further developed by R. M. Burstall and R. J. Popplestone. The language is fully defined in the reference manual (Burstall and Popplestone, 1968). The reference manual, although defining every feature of the POP-2 language, is not a suitable introduction for the beginner any more than a French-English dictionary is a suitable introduction to French.

POP-2 differs from most programming languages because it is designed to be suitable for non-numerical as well as for numerical uses. It also is a conversational language which allows the user to communicate with the system and vice versa while a job is in progress. The simplest use of a POP-2 system is as a rather powerful calculating machine in which each step in a calculation is carried out immediately after it has been requested. At the other end of the scale, it can be used as a conventional computing system in which a complete program of instructions representing the whole job is given to the system. In practice a combination of these two modes is used so that the user can choose the size of a job step that is convenient to him at a particular time. POP-2 is a powerful and sophisticated language for non-numerical calculations but is not intended for heavy duty numerical work where high efficiency is required.

The object of this manual is to introduce the reader to the POP-2 language, particularly its simpler features and terminology. Not all POP-2 facilities are described. Having read this manual, however, the reader is then in a position to develop this kernel of knowledge with the aid of the POP-2 reference manual. We have tried as far as possible to make this introduction intelligible to someone with no previous knowledge of programming.

One way to use this manual is to sit down at a POP-2 console and try out each example (and ones of your own) as it occurs in the text. It is possible to follow the text without a console but it should be kept in mind that the examples occur in sequence in each section so that if the variable x is given

the value 2 in an example in the early part of a section it keeps this value unless of course any intervening example changes it.

The notation used in this document is that of the reference manual. The rules for typing POP-2 on a particular implementation will depend upon the given implementation. These rules for the Elliott 4120 implementation are described in the Multi-POP functional specification (Collins and Pullin, 1967). The appropriate functional specification must be consulted before trying out the examples that follow.

2. ARITHMETIC AND THE STACK

The simplest use of a POP-2 system is as a rather high-powered calculating machine. If an arithmetic expression such as

$$12.0 + 2.5* (1.5 + 5/2) \Rightarrow$$

is typed, the expression is evaluated and printed thus

The print arrow '⇒' typed after an expression (or series of expressions separated by commas) causes the expression (or expressions) to be evaluated and the value (or values) to be printed. '⇒' is typed as two characters '=>'. As well as addition, subtraction, multiplication and division, mathematical functions like square root and trigonometric functions are provided. Thus

$$2.5_{10} - 2 * sqrt(3*3 + 4*4) \Rightarrow$$

causes the result

to be printed. The subscript 10 followed by a positive or negative integer scales a real number by moving the decimal point the specified number of places to the right (positive) or left (negative).

Any expressions evaluated have their results placed on the *stack*. The stack is like a stack of cards on which values are written and behaves similarly in the sense that the last item placed on the stack is the top item and is the first item to be removed from the stack. Thus if we type

$$1, 3.5 + 1.5, sqrt(2) \Rightarrow$$

the three numbers 1, 5 and 1.414 are placed on the stack with 1.414 at the top. The print arrow '=>' prints (and destroys) the whole stack starting at the bottom and hence in this case would cause

to be printed.

As an exercise, try evaluating the following expressions using the POP-2 console:

$$\frac{2 \cdot 5 * 2}{-1 \cdot 5 * 4} \qquad 1 + 2(5 - 3) \qquad \sqrt{(3^2 + 4^2)}$$

3. SIMPLE VARIABLES

Values may be assigned to *variables* for later use. Variables have identifiers consisting of up to eight letters and digits, beginning with a letter. Variables must be *declared* before use. A typical declaration is

```
vars a b sum;
```

which declares variables called a, b and sum. The list of variables declared is terminated by a semi-colon. (Although 'vars' is in heavy print here for ease of reading, it is typed in the usual characters.) Initially a variable has no value so if we print out the value of a by typing

$$a \Rightarrow$$

the result

**undef

is printed showing that the value of a is undefined.

A variable may be given a value by using an assignment. The assignment

$$2.5 \rightarrow a$$
;

results in the variable a being given the value 2.5 (' \rightarrow ' is typed as '->'). Thus after this assignment, if we type

 $a \Rightarrow$

the result

**2.5

will be printed. Expressions of any complexity may appear on the left-hand side of an assignment. Thus the sequence

```
2*2 \rightarrow a;

3*a \rightarrow b;

a*a+b*b \rightarrow sum;

sum \Rightarrow
```

will cause the result

**160

to be printed. Printing the value of a variable does not change its value.

Note that a semi-colon is used to separate each assignment from whatever follows it. Failure to do this will result in an error.

4. FUNCTIONS

If a series of similar calculations need to be done with only one or two values being changed between one calculation and the next, it is advantageous to give a name to the sequence of steps involved so that only the name has to be given for each calculation instead of the list of steps. This is, of course,

the crux of programming. Thus if we wished to evaluate a series of expressions such as

```
3·5*3·5+2·5*2·5
8·36*8·36+1·48*1·48
3·5*3·5+1·48*1·48
```

we could type out these expressions followed by a print arrow and get the result printed in each case. It would be easier and more elegant, however, to declare first a function which squares a pair of numbers and adds them together. A suitable function declaration is

```
function sumsq x y;

x*x+y*y

end
```

Having declared the function sumsq, it can be used as follows:

```
sumsq(3.5, 2.5), sumsq(8.36, 1.48), sumsq(3.5, 1.48) \Rightarrow causing the results
```

```
**18·5, 72·08, 14·44
```

to be printed. The saving would of course be much greater for a complicated function. We have now added to the list of built-in functions like *sqrt*, a function of our own. Thus the user can build up a set of functions specially designed for his own needs.

The body of a function declaration can contain several statements if necessary. The result of the function is left on the stack. For example

```
function f x y;

vars u v;

x+y \rightarrow u;

x-y \rightarrow v;

u^*u+v^*v

end
```

The variables u and v have been declared within the function declaration for use as workspace within the function. They are called *local variables*. This is because u and v are declared only for use within the function body and do not have any meaning anywhere else, thus avoiding possible interference with other (coincidental) use of u or v for variables.

The variables x and y used in this function declaration are called *formal* variables. They are used to represent the value which will later be supplied as the parameter enclosed in parentheses when the function is used. Thus

$$f(5+2, 3) \Rightarrow$$
**116

Note that when statements and expressions are typed within a function body they are not executed or evaluated immediately as they are when the system is being used as a desk calculator. Within a function body, the statements and expressions themselves are stored away ready for use when the

BURSTALL AND COLLINS

function is used. A function can be used in quite a complex way. Thus we can type

```
sumsq(6,sumsq(4,3))
which prints the result
**661
```

As an exercise, declare a function of three parameters, say a, b and c and which has as results the two values representing the roots of the quadratic equation $ax^2 + bx + c$. These can be computed from the expressions

$$\frac{-b\pm\sqrt{(b^2-4ac)}}{2a}$$

Having got this basic function working, improve it:

- 1. To take account of negative values of b^2-4ac .
- 2. To take account of a zero value for a.
- 3. To avoid doing wasteful calculations such as taking the square root of the same thing twice.

CONDITIONALS, LOOPS AND LABELS

In writing a function definition we may wish to make the outcome depend upon certain tests

```
function max \ x \ y;

if x > y then x else y

close

end

function increment \ x;

if x > 0 then x+1 \to x else x-1 \to x

close

end
```

The expressions between if and close are called *conditional expressions*. The condition may involve the operations =, <, >, = < (less than or equal) or > = (greater than or equal), and a sequence of such simple conditions separated by and's and or's may be used e.g.

if
$$x > 3$$
 and $y = < 2$ then $x + y$ else $x - y$ close

If the condition has the value true the part after then is done, otherwise the part after else is done.

The expressions after then and else may be expressions having a value such as x+3, or they may be statements or even sequences of statements e.g.

if
$$x = 0$$
 then $y+1 \rightarrow y$; $z+1 \rightarrow z$
else $y-1 \rightarrow y$; $z-1 \rightarrow z$
close

The words else and close are sufficient to terminate these sequences. We may wish to have more than one condition as in the following expression

```
if x = 1 then 2
elseif x = 2 then 3
elseif x = 3 then 4
             else ()
```

close

Note the single word elseif used to introduce the extra conditions.

Sometimes it is convenient to specify only one alternative, and we omit the else

if
$$x = 3$$
 then $0 \rightarrow y$ close

This only makes sense when the then is followed by a statement. The following does not make sense since it has no value if x is not 3

```
if x = 3 then 1 close
```

Sometimes we wish to repeat a calculation for a number of different values of a variable. This is done by using a goto statement and a label. For example to add up the squares of the numbers from 1 to n

```
function squares n; vars sum;
     0 \rightarrow sum;
loop: sum + n * n \rightarrow sum; n-1 \rightarrow n;
     if n > 1 then goto loop close;
     sum
end
```

Here sum is a local variable introduced purely for the internal working of the function squares. As we keep going back to the label loop the next three statements are repeated until eventually n = 1 and the function terminates. Note how we first calculate sum and then write it at the end to provide the result of the function.

To tabulate the values of a function f of two arguments from 1 to m and 1 to n

```
function tab f m n; vars i j;
         1 \rightarrow i:
  li: if i = < m then 1 \rightarrow j
                     lj: if j = \langle n \text{ then } f(i, j) \Rightarrow
                                 i+1 \rightarrow i; goto li
                         close:
                         i+1 \rightarrow i; goto li
        close
   end
Using the function sumsq defined previously
         tab(sumsq,2,3);
   **5
```

**10

BURSTALL AND COLLINS

**5

**8

**13

The layout could be improved by avoiding the print arrow in

$$f(i,j) \Rightarrow$$

and using instead pr(f(i,j)) which prints without a newline or asterisks. We would then get

```
tab(sumsq,2,3); 2 5 10 5 8 13
```

Better still after $1 \rightarrow i$ insert nl(1) which produces one newline

tab(sumsq,2,3);

2 5 10

5 8 13

Labels and goto statements can also be used to jump forward, i.e. to skip over a piece of program. Although it is usually clearer to do this using a conditional expression, there are occasions when the conditional would be large and clumsy. An example of using labels:

```
if x = 1 then 1 \to y else 1 \to z close
could be written (although less clearly)
if x = 1 then 1 \to y; goto on close; 1 \to z;
```

on:

Instead of using loops to repeat a piece of program we may use a device known as recursion, i.e. a function which calls itself.

We can define the factorial function

by looping:

by recursion:

```
function fact n;

vars p; 1 \rightarrow p;

loop: if n = 0 then p

else n * p \rightarrow p; n-1 \rightarrow n

goto loop

function fact n;

if n = 0 then 1

else n * fact(n-1)

close
```

end

ARRAYS

So far, variables have been used only to denote single values. It is sometimes useful to give a name to an array of several values. The individual component values of an array are extracted by means of one or more integer subscripts.

The sequence

```
vars a;
```

 $newarray([\% 1, 10, 1, 20 \%], sumsq) \rightarrow a;$

declares a variable a and makes it into an array with two dimensions. The elements of the array a are accessed by means of two integer subscripts; the

first in the range 1 to 10 and the second in the range 1 to 20. The decorated brackets [% and %] are used simply to delimit the subscript list. The value of each element a(i,j) is initially set to the corresponding value of sumsq(i,j). Thus for integer values of i and j within their appropriate ranges, a(i,j) is not distinguishable from sumsq(i,j). However, because the individual values of a are individually stored, any one may be assigned a new value without affecting the value of a(i,j) for other values of i and j. Thus a(2,3) initially has the same value as sumsq(2,3) but the assignment

$$10 \to a(2, 3);$$

gives a(2, 3) the new value of 10 without affecting the value of a(i, j) for other values of i and j.

The similarity between functions and arrays is often extremely useful. For example, if a function for multiplying arrays together is being used, it will accept a function as well as arrays. The diagonal array

```
1 0 0 0
0 1 0 0
0 0 1 0
0 0 0 1
```

is much better represented by the function

```
function diag \ i \ j;

if i = j then 1 else 0 close

end
```

than an array which would take up more space, particularly if the matrix is large.

An example of a function to do matrix addition:

```
function zero i j; 0 end

function matadd m1 m2 ni nj; vars i j m3;

1 \rightarrow i; newarray([\%, 1, ni, 1, nj \%], zero) \rightarrow m3;

iloop: if i = < ni then 1 \rightarrow j;

jloop: if j = < nj then m1(i, j) + m2(i, j) \rightarrow m3(i, j);

j+1 \rightarrow j; goto jloop

close;

i+1 \rightarrow i; goto iloop

close;

m3

end:
```

Use this function to add together the matrices

$$\begin{bmatrix} 1 & 3 \\ 2 & 6 \\ 3 & 9 \end{bmatrix} \qquad \text{and} \qquad \begin{bmatrix} -4.5 & 0.36 \\ 2.51 & -0.1 \\ 6.24 & 5.09 \end{bmatrix}$$

with the first matrix represented first by a function and secondly by an array.

BURSTALL AND COLLINS

It would be much easier to define matadd as follows

function matadd m1 m2 ni nj;

```
function sum i j; m1(i, j) + m2(i, j) end; newarray([% 1, ni, 1, nj %], sum)
```

The function *sum* is here defined locally within *matadd* and only mentioned once. We could avoid giving it a name and a separate definition as follows:

function matadd m1 m2 ni nj; newarray([%1, ni, 1, nj %], lambda i j; m1(i, j)+m2(i, j) end) end

The lambda is used to denote an anonymous function.

lambda ij; ... end means 'the function of i and j whose definition is ...'. Thus lambda replaces the word function and the name of the function. Compare the function called 'sum' with a number called 'three'

$$1+2 \rightarrow \text{three};$$

 $4 + three \Rightarrow$

We could do without the name 'three', saying instead

$$4+(1+2) \Rightarrow$$

The latter case is like a **lambda** expression, which is an anonymous function just as (1+2) is an anonymous number.

7. LISTS AND NON-NUMERICAL VALUES

Programs do not only deal with numbers, and POP-2 allows items called words. These look like identifiers but do not denote a variable, i.e. they have no value. We use quotes to distinguish them from variables. Compare

vars
$$cost$$
; $8 \rightarrow cost$; $cost \Rightarrow$ **8

with

$$"cost" \Rightarrow **cost$$

The list of expressions separated by commas and enclosed in decorated brackets [% ... %] has already been introduced in the section on POP-2 arrays.

The sequence

vars
$$i, j;$$

 $1 \rightarrow i; 2 \rightarrow j;$
 $[\%, i, i+j, "dog", "cat" \%] \rightarrow u;$

assigns the list of values

1 3 dog cat

to the variable u. Note that each item in the list is evaluated before the assignment takes place. When all the items in a list are words or positive integers instead of writing

$$[\%"dog","cat", 3, 57 \%]$$

we can use the brackets '[' and ']' and write more briefly

instead. These brackets behave in the same way as decorated brackets within which all words are enclosed in quotes. Note that undecorated brackets need no commas and may only contain words and unsigned numbers.

The first item in a list, whether it is a single item or itself a list is known as the *head* of the list. The list remaining after the head of a list has been removed is known as the *tail* of the list. The standard functions hd and tl are used to split lists in this way. Thus as u is the list $\lceil 1 \rceil$ 3 dog cat \rceil , typing

$$hd(u) \Rightarrow$$

results in

**1

being printed and typing

$$tl(u) \Rightarrow$$

results in

being printed. A function cons enables an item to be added before the head of a list so that

$$cons("pig", u) \Rightarrow$$

results in

being printed. The operation :: can be used instead of *cons*; the previous request could have been typed as "pig'' :: $u \Rightarrow$ instead of $cons("pig", u) \Rightarrow$

There is another operation which joins lists together. The sequence

$$[a \ b \ c] \langle \rangle \ u \ \langle \rangle [x \ y \ z] \Rightarrow$$

results in

**[
$$a b c 1 3 dog cat x y z$$
]

being printed. We could use this joining operation to define a function similar to *cons* which adds items to the end rather than the beginning of a list. The declaration

function append
$$x y$$
;

$$y \Leftrightarrow [\% x \%]$$

end

defines such a function. Having defined append, typing

append
$$(4, [\%1, i+1, 3\%]) \Rightarrow$$

BURSTALL AND COLLINS

results in

being printed.

Note that [% x %] is not the same as x. It is a list of one item. We sometimes want an empty list, a list of no items. This is called *nil*, also written []. Thus

```
3 :: nil \Rightarrow **[3]
```

We can recognise an empty list using the function *null*. This is useful if we want to process all items of a list. For example to find the largest element of a list

```
function largest xs; vars x max;

-1000000 \rightarrow max;

loop: if null(xs) then max

else hd(xs) \rightarrow x; tl(xs) \rightarrow xs;

if x > max then x \rightarrow max close;

goto loop

close

end;

largest([1\ 3\ 5\ 2]) \Rightarrow
**5
```

As an exercise, define an analogous function for finding the smallest item in a list.

8. RECORDS

Records are similar to arrays in that a number of component values are associated with a single name. Unlike arrays, however, all records of a particular record class must have the same number of components and these components are referred to by name rather than by using a numeric subscript.

Having declared some variables:

```
vars consper destper forename surname male p1 p2;
```

the standard function *recordfns* can be used to set up a record class and give us some useful functions for dealing with records within this record class. Thus

```
recordfns("person", 500,[0 0 1]) \rightarrow male \rightarrow surname \rightarrow forename \rightarrow destper \rightarrow consper;
```

sets up a class of records called "person" for handling facts about a number of people, say 500 or so. Each person has a forename, a surname and a maleness. The maleness is either 0 (false) or 1 (true). The list $[0\ 0\ 1]$ shows the size of each item; 0 means any kind of item, and a positive number k means an integer n such that $0 < n < 2^k$ (hence for maleness k = 1 is sufficient). The function consper creates a person from 3 suitable components and destper decomposes a person into 3 components.

```
To construct two persons:
     consper("john", "smith", true) \rightarrow p1;
     consper("jane", "jones", false) \rightarrow p2;
To examine them:
forename(p1) \Rightarrow
     **john
     surname(p2) \Rightarrow
     **jones
     male(p1) \Rightarrow
     **1
Note: 1 is used to represent true. To change a person:
     surname(p1) \rightarrow surname(p2);
     surname(p2) \Rightarrow
     **smith
To decompose one of them:
     vars f s m; destper(p2) \rightarrow m \rightarrow s \rightarrow f;
     f \Rightarrow
     **jane
     s \Rightarrow
     **smith
     m \Rightarrow
     **0
   A function to marry a girl:
     function marry boy girl;
          if not(male(boy)) or male(girl) then pr("illegal")
                    else surname(boy) \rightarrow surname(girl)
          close
end;
```

As an exercise in the use of arrays and records, write a group of functions for maintaining an array of records describing people say giving name, sex and age. Functions should include

addname to add a new name with properties

delname to delete an existing name

to print all properties of all names printall

to increase the age of a person with given name by one

Printall can be refined to list selected properties for selected names. The selection of names might depend upon whether some condition is true or not for the given name.

INPUT AND OUTPUT FACILITIES

We have seen already how output can be printed using the ⇒ symbol which prints and empties the contents of the stack from the bottom upwards.

BURSTALL AND COLLINS

Other standard functions for input and output are:

```
itemread ()reads one simple item producing its value as result.pr(x)outputs one item x.print(x)outputs one item x and leaves x on the stack.nl(n)causes further output to continue on a new line after n-1 blank lines.sp(n)skips n spaces across the page.
```

For example, a function to read a list of numbers terminated by the word 'end' and print the total might be defined as follows:

```
function sum;

vars x total; 0 \rightarrow total;

kl: itemread ( ) \rightarrow x;

if x = "end" then pr (total);

else total + x \rightarrow total; else total
```

After calling the function sum, the numbers to be added are typed in. The total is printed when the word end is typed.

In addition to the above functions, individual characters may be read and output using the functions *charin* and *charout*. Characters are represented by integers, the actual correspondence depending upon the particular implementation.

Facilities will be provided on most installations to read and output information on devices other than the on-line console. The standard function popmess when given a list whose head is the name of a device and whose tail is a file name produces as value a function for reading or outputting characters, whichever is appropriate for the particular device. Device names will vary from one installation to another. The examples below use the names ptin, ptout and lp80 to refer to a paper tape reader, paper tape punch and line-printer respectively. For example, to print up a paper tape file called [abc prog2] on the line printer labelling the printed output as [abc prog3] we would type:

```
vars readch printch;

popmess ([ptin abc prog2]) \rightarrow readch;

popmess ([lp80 abc prog3]) \rightarrow printch;

function list; vars c;

li: readch () \rightarrow c; printch (c); if not (c=termin) then goto li close

end

list ();
```

The first call of *popmess* results in the named paper tape file being loaded on a paper tape reader and produces as a result a function which we assign to *readch* for reading successive characters from the paper tape file. The second call reserves the line-printer, outputs the file name and produces a

function for outputting characters to the line-printer. This function is assigned to *printch*. The function *list* simply reads and prints each character in turn and continues to do so until the input file terminates.

It is also possible to make the system itself read from a device other than the POP-2 console. For example,

```
vars a;
popmess ([ptin xyz prog4]) \rightarrow a;
compile (a);
```

is exhausted.

causes the system to continue reading and executing POP-2 text from the paper tape file labelled [xyz prog4]. The function compile could be defined as

function compile a; fntolist (incharitem (a)) <> proglist \rightarrow proglist end Further input continues from the on-line console when the paper tape file

The output produced by any program using the standard functions nl, sp, pr and print can be redirected to an alternative channel by assigning a character output function to the standard variable cucharout. Thus:

```
popmess([ptout\ abc\ output]) \rightarrow cucharout;
```

would result in output being produced on a paper tape punch until further notice.

10. ADDITIONAL EXERCISES

- 1. Write a function to print a table of numbers with their squares, cubes and reciprocals from 1 to n.
- 2. Write a function to compute e^x as $1+x+x^2/2!+x^3/3!$... continuing until the terms are less than 001. Check that $e^x e^y = e^{x+y}$ to the expected accuracy.
- 3. Write a function to find the average of an array of numbers. Write a function to print all numbers in an array which are more than 3 times the average value of the array.
- 4. Write a function to test whether a 2 dimensional array of integers is a magic square (all rows, columns and diagonals have the same total and every number from 1 to n is present).
- 5. Write a function to take an array whose elements are the words 'nought', 'cross' or 'blank' and test whether it is a winning position in noughts and crosses. Write a function to put a nought in a square which creates a win for nought if possible, or if not in one which prevents a win for cross if possible, or otherwise in an arbitrary square.
- 6. Write a function to count how many items in a list of integers are greater than n. Write a function to count how many have the property p, where p is a given function which produces a truth value as its answer.
- 7. Write a function to convert a list of pairs into a pair of lists e.g. to

BURSTALL AND COLLINS

convert [[dog bitch] [cock hen] [drake duck]] into [[dog cock drake [bitch hen duck]].

8. Write a function 'edit' to delete from the middle of a list a given list and replace it with another e.g.

edit([Jim is a son of a bitch and so is Bob], [son of a bitch], [* * * *])

= $\lceil Jim \ is \ a * * * * and so \ is \ Bob \rceil$.

9. A point is a pair of real numbers.

A triangle is a point and a point and a point.

Create suitable functions for constructing and manipulating points and triangles, and write a function to test whether a triangle is equilateral.

10. An 'expression' is either an integer or a compound which has 3 components, a function, an expression and an expression (the functions may be any function of 2 arguments).

Create functions to construct and manipulate expressions and write a function to find the value of an expression.

11. A dictionary is a chain of 3 component records thus:

		1	 ·						
CAT	CHAT		 DOG	CHIEN	-		COW	VACHE	NIL
0111	CIMIA	L	200			l			

Write a function to add a pair of items to a dictionary producing a new dictionary and a function to look up an item in a dictionary.

11. ADVANCED FEATURES

A number of features of POP-2 have been omitted from this introduction. Two important ones are 'partial application' and 'dynamic lists'. These are described in the reference manual but it might be worth while explaining their uses briefly. Further explanation and examples may be found in Popplestone (1968a).

Partial application is a device for obtaining from a given function another function with fewer formal parameters, the missing parameters taking on fixed values. Suppose that we have a function distance(u,v,x,y) and we often want to use distances from the origin point (0,0). We could define

function $dist0 \ u \ v$; distance(u,v,0,0) end

In POP-2 we can say simply

 $distance(\%0, 0\%) \rightarrow dist0;$

This 'freezes' the last two parameters of distance. The function dist0 carries the values (0,0) for these parameters around with it.

Another use for partial application is where we want to use a function with some free variables at a time when these variables have assumed other values or are not defined at all e.g. the function f defined as

function
$$f x$$
; $x+a$ end

which has a as a free variable. We can ensure that a always keeps the value it had when the function was defined (not the current value when the function

is used) by making it a parameter and then freezing it thus

function $f \times a$; x+a end; $f(\% a \%) \rightarrow f$;

This gives a variant of the facility called a 'closure' devised by P. J. Landin (1964).

Dynamic lists are a device to make use of long perhaps infinite lists without using much store. They are lists some of whose values are obtained by calculating rather than being stored. Indeed the values may be read in so that we can represent a paper tape reader as a dynamic list.

Normally a list finishes with the 'empty list' nil in its tail. Instead of this we may insert a function of no arguments which may be called repeatedly. This is used to calculate the remaining items in the list. The standard functions hd and tl are so adjusted that by taking its tl repeatedly one may process a list without knowing whether it is partly or wholly dynamic. Moreover the values calculated are chained on to the end of the list so that they may be reaccessed when required. If they are not required the storage control scheme will eventually collect the space for re-use.

This gives a variant of the facility called a 'stream' devised by P. J. Landin (1965).

REFERENCES

- Burstall, R. M., and Popplestone, R. J. (1968). POP-2 reference manual. *Machine Intelligence 2*, 205-44, also this volume, eds. E. Dale and D. Michie. Edinburgh: Oliver and Boyd.
- Collins, J. S., and Pullin, D. J. S. (1967). POP-2 functional specification. *Mini-MAC Reports: No. 1*, Department of Machine Intelligence and Perception, University of Edinburgh.
- Landin, P. J. (1964). The mechanical evaluation of expressions. *Computer Journal*, 6, 4, 308-20.
- Landin, P. J. (1965). A correspondence between ALGOL 60 and Church's lambda notation. Part I. Communs. Ass. Comp. Mach., 8, 2, 89-101.
- Popplestone, R. J. (1968). POP-1: an on-line language. *Machine Intelligence* 2, 185-194, eds. E. Dale and D. Michie. Edinburgh: Oliver and Boyd.
- Popplestone, R. J. (1968a). The design philosophy of POP-2. *Machine Intelligence 3*, ed. D. Michie. Edinburgh: Univ. Press.

動しても April And App はかからからからためにきない場合が

POP-2 REFERENCE MANUAL

BY

R. M. BURSTALL

AND

R. J. POPPLESTONE

DEPARTMENT OF MACHINE INTELLIGENCE AND PERCEPTION UNIVERSITY OF EDINBURGH

ADDENDA

These alterations are subsequent to the publication of Machine Intelligence 2.

Errata for POP-2 Reference Manual

- Section 2.2. Explanation of //, for ' $0 \le r < b$ ' read '|r| < |b| and $r * a \ge 0$ '.
- Section 2.3. For 'there are also operations' read 'There are also functions'.
- Section 4.1. In definitions of formal parameter list element and output local list element, for 'ident' read 'identifier'. In definition of lambda expression, after 'output local list' insert '?'.

In last line of page for '(lambda', read 'lambda'.

- Section 5.5. In definition of quasi compound expression, after 'dot operator *' delete '?'.
- Section 6.1. Last line, for 'elseif' read 'else if'.
 Section 8.3. In definition of solidified, for '1' read 'l' throughout. After example of list constant, for 'list expressions are formed' read 'list expressions are evaluated'.
- Section 8.5. In 'the boundslist is a list of integers, these two' delete 'two'.
- Section 9.1. In definition of incharitem, for '(1) textitem' read '() > textitem'.

Amendments to POP-2 Reference Manual

Use of the language has suggested the following minor revisions.

- 1. The function *intof* should be made the same as Algol *entier*. Section 2.3. Explanation of intof, after 'to the nearest integer' add 'not greater than it'.
- 2. It is more convenient for a destructor function not to delete the data structure.
 - Section 7.1. First page, delete the paragraph 'After applying the destructor to a structure it is deleted.'
 - Section 8.3. Replace the definition of dest by 'function dest 1; hd (1), tl(l) end'.
 - Last paragraph delete the line beginning 'next ϵ ...'.

CONTENTS

_	INTRODUCTION		PAGE
1	INTRODUCTION		3
	1.1 Aims		
	1.2 Main features		
	1.3 Examples		-
	1.4 Notation for syntactic description		3 4 6 7
	1.5 Notation for functions		′
2	ITEMS		
	2.1 Simple and compound items		8
	2.2 Integers		9
	2.3 Reals		9
	2.4 Truth values		10
	2.5 Undefined		10
	2.6 Terminator		10
3	VARIABLES		
Ī	3.1 Identifiers		10
	3.2 Declaration and initialisation		11
	3.3 Cancellation		13
4	FUNCTIONS		
•	4.1 Definition of functions		13
	4.2 Application of functions		14
	4.3 Nonlocal variables		15
	4.4 Partial application		15
	4.5 Doublets		17
	4.6 Arithmetic operations		18
5	EXPRESSIONS AND STATEMENTS		
_	5.1 Expressions		18
	5.2 Precedence	+	20
	5.3 Statements and imperatives		20
	5.4 Labels and goto statements		21
	A 4 1		

CONTENTS

	5.5 Assignment		PAGE 22
	5.6 Comments		23
6	CONDITIONALS		
	6.1 Conditional expressions		23
	6.2 Conjunctions and disjunctions		24
7	DATA STRUCTURES		
	7.1 Functions of data structures		25
	7.2 Records		27
	7.3 Strips		28
	7.4 Garbage collection		29
8	STANDARD STRUCTURES		
	8.1 References		30
	8.2 Pairs		30
	8.3 Lists		30
	8.4 Full strips and character strips		33
	8.5 Arrays		33
	8.6 Words		34
	8.7 Functions		35
9	INPUT AND OUTPUT		
	9.1 Input		36
	9.2 Output		37
77.	ta 🍷	•	٠,
10	MACHINE CODE		38
11	MODES OF EVALUATION		
	11.1 Immediate evaluation		38
	11.2 Macros	The state of the s	38
	11.3 Evaluation of program text		39
	. •		-
	Acknowledgments		39

1. INTRODUCTION

1.1. Aims

The following are the main design objectives for the POP-2 language:

- (i) The language should allow convenient manipulation of a variety of data structures and give powerful facilities for defining new functions over them.
- (ii) The language should be suitable for taking advantage of on-line use at a console, i.e. it should allow immediate execution of statements and should have a sufficiently simple syntax to avoid frequent typing errors.
- (iii) A compiler and operating system should be easy to write and should not occupy much storage.
- (iv) The elementary features of the language should be easy to learn and use.
- (v) The language should be sufficiently self-consistent and economical in structure to allow it to incorporate new facilities when extensions are desired.

In attaining these objectives certain other desirable features of programming languages had to be relegated to secondary importance:

- (vi) Fast arithmetical facilities on integer and real numbers.
- (vii) Fast subscripting of arrays.
- (viii) A wide variety of elegant syntactic forms.

Naturally whether (iii) or (vi) and (vii) are attained is to a considerable extent a matter of implementation.

1.2. Main features

The following main features are provided. Roughly analogous features of some other programming languages are mentioned in brackets as a guide:

- (i) Variables (cf. ALGOL but no types at compile time).
- (ii) Constants (cf. ALGOL numeric and string constants, LISP atoms and list constants).

- (iii) Expressions and statements (cf. ALGOL).
- (iv) Assignment (cf. ALGOL, also CPL left-hand functions).
- (v) Conditionals, jumps and labels (cf. ALGOL but restrictions on jumps and labels).
- (vi) Functions (cf. ALGOL procedures but no call by name, cf. CPL and ISWIM for full manipulation of functions).
 - (vii) Arrays (cf. ALGOL; cf. CPL for full manipulation of arrays).
 - (viii) Records (cf. COBOL, PL/1, Wirth-Hoare ALGOL records, CPL nodes).
 - (ix) Words (cf. LISP atoms).
 - (x) Lists (cf. LISP, IPL-V).
 - (xi) Macros.
 - (xii) Use of compiler during running (cf. LISP, TRAC, FORMULA ALGOL).
 - (xiii) Immediate execution (cf. JOSS, TRAC).

Notes:

LISP: LISP 1.5

CPL: See Barron, D. W., et al. 1964. The main features of CPL, Computer J., 6, 134-43.

CPL reference manual. Edited C. Strachey (privately circulated).

Wirth-Hoare ALGOL: See Wirth, N., and Hoare, C. A. R. 1966. A contribution to the development of Algol, *Communs Assn Comput. Mach.*, 9, 413-32.

TRAC: See Mooers, C. N. 1966. TRAC, a procedure describing language for the reactive typewriter, Communs Assn Comput. Mach., 9, 215-24.

ISWIM: See Landin, P. J. 1966. The next 700 programming languages, Communs Assn Comput. Mach., 9, 157-166.

1.3. Examples

The following is an example of POP-2 program text. The sign \Rightarrow (not to be confused with that used in section 1.5 'Notation for functions') prints out some results on a newline prefixed with two asterisks. These results are included in the text below, as they would appear if the program were run on-line at a console.

```
comment arithmetic;

12 \cdot 0 + 2 \cdot 5*(1.5 + 2.5) \Rightarrow

**22.0

vars a b sum;

2*2 \rightarrow a; \ 3*a \rightarrow b; \ a*a + b*b \rightarrow sum; \ sum \Rightarrow

**160

function sumsq x y;

x*x + y*y

end;

sumsq(a, b) + 1 \Rightarrow

** 161
```

```
BURSTALL AND POPPLESTONE
function fact n; vars p;
      1 \rightarrow p;
loop: if n=0 then p else n*p \rightarrow p; n-1 \rightarrow n; goto loop close
end:
fact(fact(3)) \Rightarrow
** 720
comment arrays;
vars a i j;
10 \rightarrow i; 20 \rightarrow j;
newarray([\%1, i, 1, j\%], sumsq)\rightarrow a;
a(2, 3) \Rightarrow
** 13
10 \rightarrow a(2, 3); a(2, 3) \Rightarrow
** 10
function arraysum a1 a2 m n;
     newarray ([%1, m, 1, n%], lambda i j; a1(i, j) + a2(i, j) end)
end:
arraysum (a, a, 10, 20) \rightarrow a; a(2, 3) \Rightarrow
** 20
comment lists;
vars u;
1 \rightarrow i; 2 \rightarrow j;
[\%i, i+j, "dog", "cat" \%] \rightarrow u; u \Rightarrow
** [1 3 dog cat]
cons("pig", u) \Rightarrow
** [pig 1 3 dog cat]
function append x y;
      if null (y) then [\% x \%] else cons(hd(y), append(x, tl(y))) close
append(4, [\% 1, i+1, 3\%]) \Rightarrow
** [1 2 3 4]
comment records;
vars consper destper forename surname male p1 p2;
```

vars consper destper forename surname mule p1 p2; recordfns("person", 500, [0 0 1])→male→surname→forename →destper→consper;

consper("jane", "jones", false) $\rightarrow p1$; consper("sam", "smith", true) $\rightarrow p2$;

 $surname(p1) \Rightarrow$

** jones $datalist(p1) \Rightarrow$

** [jane jones 0] routine marry x y;

if male(x) and not(male(y)) then $surname(x) \rightarrow surname(y)$ close

end;

marry(p2, p1); $datalist(p1) \Rightarrow$ ** [jane smith 0]

1.4. Notation for syntactic description

We use the BNF (Backus-Naur Form) notation as used in the ALGOL report:

```
::= indicates a syntax definition;

< > are used to enclose the name of a syntax class;
| denotes disjunction (union of syntax classes).
```

Concatenation denotes concatenation of any elements of two syntax classes.

We also use a convenient extension of this notation due to R. A. Brooker:

- * means that a class may occur n times, $n \ge 1$:
- ? means that a class may occur n times, n=0 or 1;
- *? means that a class may occur n times, $n \ge 0$.

e.g. the definitions

```
\langle astring \rangle := \langle a \rangle \langle astring \rangle | \langle a \rangle
\langle bstring \rangle := \langle b \rangle \langle astring \rangle
\langle cstring \rangle := \langle c \rangle \langle astring \rangle | \langle c \rangle
```

may be replaced by

$$\langle bstring \rangle := \langle b \rangle \langle a* \rangle$$

 $\langle cstring \rangle := \langle c \rangle \langle a*? \rangle$

The characters \langle , \rangle and * are used in the POP-2 reference language but no confusion should arise.

When we wish to give examples of a syntax class we use the symbol 'e.g.::=', for example:

$$\langle bstring \rangle e.g. := \langle b \rangle \langle a \rangle | \langle b \rangle \langle a \rangle \langle a \rangle \langle a \rangle$$

The character set of the POP-2 reference language is as follows.

```
\langle letter \rangle ::= a|b|c|d|e| f|g|h|i|j|k|l|m|n|o|p|q|r|s|t|u|v|w|x|y|z
\langle digit \rangle ::= 0|1|2|3|4|5|6|7|8|9
\langle sign \rangle ::= + |-|*|/|$|\&| = |\langle|\rangle|:|£|\uparrow
\langle separator \rangle ::= ,|;
\langle period \rangle ::= .
\langle sub\ ten \rangle ::= _{10}
\langle bracket \rangle ::= (|)|[|]
\langle bracket\ decorator \rangle ::= %
\langle quote \rangle ::= "
\langle string\ quote \rangle ::= \langle || \rangle
```

BURSTALL AND POPPLESTONE

Letters may be written in lower case, upper case or heavy type without any change of meaning. It will be conventional however to use heavy type letters for syntax words, i.e. those identifiers such as function, then, end and cancel which have a special meaning for the POP-2 compiler and which characterise certain syntactic forms.

Spaces, tabulate and new lines terminate identifiers, integers, reals and words but otherwise they are ignored.

A distinction is made between the reference language used in this document and a number of possible hardware languages used by particular computer implementations of POP-2. Each character in the reference language should be represented by a distinct character or sequence of characters in the hardware language. A particular letter, whether upper case, lower case, heavy type or not is regarded as the same character in the reference language.

The symbols \rightarrow and \Rightarrow used in this paper should be read as a typographical abbreviation for the pairs of characters -> (minus greater than) and => (equals greater than) respectively.

1.5. Notation for functions

It is convenient to have a notation to specify the domain and range of functions. We will consider functions having several arguments (or possibly none) and producing several results (or possibly none), the notion of functions with more than one result being an extension of normal mathematical usage (see section 4.2 'Application of functions'). We introduce a special symbol '⇒' which is not to be confused with any identifier in the POP-2 language.

Suppose d1, d2, ..., dm and r1, r2, ..., rn are all sets of items. Then $d1, d2, ..., dm \Rightarrow r1, r2, ..., rn$ is the set of all functions whose domain is d1, d2, ..., dm and range r1, r2, ..., rn, i.e. with arguments which are m-tuples in $d1 \times d2 \times ... \times dm$ and with results which are n-tuples in $r1 \times r2 ... \times rn$. We express the fact that a function f is a member of this set of functions by

$$f \in d1, d2, \dots, dm \Rightarrow r1, r2, \dots, rn$$

Some examples will make this clear.

 $add \in integer$, $integer \Rightarrow integer$ $divrem \in integer$, $integer \Rightarrow integer$, integer

where divrem is 'divide with remainder', e.g. divrem (7, 3)=2, 1 and divrem (14, 4)=3, 2

 $roundup \in real \Rightarrow integer$ $prime \in integer \Rightarrow truthvalue$

If the function has no results we use an empty pair of parentheses, thus: $printout \in integer \Rightarrow 0$

The arguments or results may themselves be functions

 $differentiate \in (real \Rightarrow real) \Rightarrow (real \Rightarrow real)$

Where we wish to discuss a number of functions all having the same domain and range it is convenient to abbreviate thus:

$$f, g, \dots h \ all \in \dots \Rightarrow \dots$$
for
 $f \in \dots \Rightarrow \dots$
and
 $g \in \dots \Rightarrow \dots$
and
 $h \in \dots \Rightarrow \dots$

Some functions do not have a fixed number of arguments and some do not have a fixed number of results (see section 4.2 'Application of functions'). In such cases we may write for example

 $f \in integer \Rightarrow real, integer, ..., integer$

for the domain or range, meaning that a real and a variable number of integers are the results.

2. ITEMS

2.1. Simple and compound items

The objects on which one can operate are called *Items*. They are divided into two distinct classes: *Compound* items, which are represented by addresses and *Simple* items which are directly represented by bitstrings which do not contain addresses (these bit strings are normally of fixed length for a given implementation, being a single machine word). The address representing a compound item points to a bit string whose length may vary from item to item. This bit string may contain other items. The areas of store immediately pointed to by two different compound items do not overlap.

The following standard function recognises compound items:

 $iscompnd \in item \Rightarrow truthvalue$

Two kinds of simple item are distinguished: integers and reals. The following standard functions recognise them:

isinteger, isreal all \in item \Rightarrow truthvalue

The standard function = (an operation of precedence 7) is used to represent equality of items. For integers and reals it has the usual meaning. Its meaning for compound items is given in section 7.1 'Functions of data structures'.

 $= \in item, item \Rightarrow truthvalue$

2.2. Integers

Integers are simple items. They may be positive, negative or zero. The size of the largest and smallest integers allowed depends on the implementation. The following functions on integers are standard:

```
intadd, intsub, intmult, all \in integer, integer \Rightarrow integer \# \in integer, integer \Rightarrow integer, integer intplus, intminus all \in integer \Rightarrow integer. intsign \in integer \Rightarrow integer integer, intle, intgreq, intleeq all \in integer, integer \Rightarrow truthvalue
```

intadd, intsub, intmult and intdiv are the usual add, subtract and multiply. $/\!\!/$ is divide with remainder and produces a quotient and a remainder (if $a/\!\!/b$ is (q, r), then q*b+r=a and $0 \le r < b$). It is an operation of precedence 4.

intplus carries an integer into itself and intminus complements an integer. intsign produces -1, 0, or +1 according to the sign of the integer. The remaining four functions are the relations 'greater than', 'less than', 'greater than or equal to' and 'less than or equal to'.

The syntax of integers is:

```
\begin{tabular}{ll} $\langle integer \rangle ::= \langle octal \ integer \rangle | \langle decimal \ integer \rangle \\ & \langle octal \ integer \rangle ::= 8 : \langle octal \ digit* \rangle \\ & \langle binary \ integer \rangle ::= 2 : \langle binary \ digit* \rangle \\ & \langle decimal \ integer \rangle ::= \langle digit* \rangle \\ & \langle octal \ digit \rangle ::= 0 | 1 | 2 | 3 | 4 | 5 | 6 | 7 \\ & \langle binary \ digit \rangle ::= 0 | 1 \end{tabular}
```

Example:

```
\langle integer \rangle e.g. := 8:777|2:101|0|6559
```

Integers may also be treated as bit-strings (the length depending on the implementation) and the following functions are standard:

```
logand, logor, logshift all \in integer, integer \Rightarrow integer lognot \in integer \Rightarrow integer
```

logand and logor are the usual bit by bit 'and' and 'inclusive or'; logshift causes the first integer to be shifted left by the number of places given in the second, unless the second integer is negative when shifting to the right takes place (all new bits to fill up the end are zero in each case).

2.3. Reals

Reals are simple items. They may be positive, negative or zero. The size of the largest and smallest reals allowed and the precision depends on the implementation. The following functions on reals are standard:

```
realadd, realsub, realmult, realdiv all \in real, real \Rightarrow real realplus, realminus all \in real \Rightarrow real realsign \in real \Rightarrow integer realgr, realle, realgreq, realleeq all \in real, real \Rightarrow truthvalue
```

These are the usual add, subtract, multiply and divide on reals. realplus carries a real into itself and realminus complements a real. realsign produces -1, 0 or +1 according to the sign of the real. The remaining four functions are the relations 'greater than', 'less than', 'greater than or equal to' and 'less than or equal to'.

There are also operations to convert a real to the nearest integer and to convert an integer to real:

 $intof \in real \Rightarrow integer$ $realof \in integer \Rightarrow real$

The syntax of reals is as follows:

 $\langle real \rangle ::= \langle decimal \ integer ? \rangle. \langle decimal \ integer \rangle \langle exponent ? \rangle$ $\langle exponent \rangle ::= _{10} + \langle integer \rangle |_{10} - \langle integer \rangle |_{10} \langle integer \rangle$

Example:

 $\langle real \rangle e.g. := .5|1.99|1.5_{10} - 6$

2.4. Truth values

The two items *True* which is the integer 1 and *False* which is the integer 0 are called *Truthvalues*.

On entry to the POP-2 system the standard variable *true* is set to 1 and the standard variable *false* is set to 0. The following standard functions on truthvalues are provided:

booland, boolor all \in truthvalue, truthvalue \Rightarrow truthvalue not \in truthvalue \Rightarrow truthvalue

These are the usual functions 'and', 'inclusive or' and 'not' of propositional calculus.

2.5. Undefined

The standard variable *undef* has the word "undef" as its value on entry to the POP-2 system (see section 8.6 'Words'). The programmer may use it as the result of a function which fails to produce its normal result.

2.6. Terminator

The standard variable *termin* has the word "termin" as its value on entry to the POP-2 system (see section 8.6 'Words'). It may be used as the first argument of a variadic function (see section 4.2 'Application of functions') or to mark the end of an input file (see section 9.1 'Input').

3. VARIABLES

3.1 Identifiers

An item may be the *Value* of a *Variable* (a variable is not itself an item). An *Identifier* is associated with the variable and this identifier is used to

BURSTALL AND POPPLESTONE

refer to it in a POP-2 program. A number of distinct variables may have the same identifier, but only one of them is *Currently associated* with it at a particular time in the evaluation process.

An identifier may be restricted to a certain range of values and it may be given special syntactic properties by being given a precedence (see section 5.2 'Precedence').

The syntax of identifiers is:

```
\langle identifier \rangle ::= \langle letter \rangle \langle alphanumeric *? \rangle | \langle sign * \rangle \langle alphanumeric \rangle ::= \langle letter \rangle | \langle digit \rangle
```

Example:

```
\langle identifier \rangle e.g. := x \mid y99 \mid alpha \mid u2a \mid +++ \mid /+ \mid \langle \rangle \mid *  * *
```

Syntax words such as then, end, \rightarrow and : have special meanings and may not be used as identifiers. Only the first 8 characters are significant.

3.2. Declaration and initialisation

A variable is either Global, Local or Formal. A Declaration is used to introduce an identifier and associate it with a global or local variable. A Local Declaration, introducing a local variable, is a declaration which occurs in a function body. A Global Declaration, introducing a global variable, is one which does not.

An *Initialisation* is used to introduce an identifier and associate it with a formal variable and give the variable an initial value. It is achieved by including the identifier in the formal parameter list of a function (*see* section 4.1 'Definition of functions').

A declaration or initialisation may also specify that the identifier is restricted to take only functions as values. This is not necessary but may make the implementation more efficient. A declaration or initialisation may also specify that the identifier is an *Operation*, i.e. it is restricted to take functions as its values and is given a precedence. This restriction is associated with the unique name (see below) produced by the declaration or the initialisation.

The syntax of declarations is:

```
\langle declaration \rangle ::= vars \langle declaration \ list \ element \ * \rangle
\langle declaration \ list \ element \rangle ::= \langle identifier \rangle \ | \langle restriction \rangle
\langle restriction \rangle ::= \langle restrictor \rangle \langle identifier \rangle \ | \langle restrictor \rangle \langle identifier \ * \rangle)
\langle restrictor \rangle ::= function \ | \ operation \langle integer \rangle
```

Example:

```
\langle declaration \rangle e.g. :=  vars x y \mid  vars x y \mid  function (fg) \mid  operation  7 =
```

A declaration or initialisation has a *Scope*, which is a piece of POP-2 text. An identifier may not be used to represent a variable outside the scope of a declaration or initialisation of the identifier.

The scope of a global declaration starts at the declaration and continues until the identifier is cancelled.

The scope of a local declaration starts at the declaration and continues to the end of the innermost function body enclosing it.

The scope of an initialisation is the body of the function in which it occurs. Each declaration or initialisation gives rise to a unique mark and this mark is associated with all occurrences of any identifier introduced by the declaration or initialisation within the scope of the declaration or initialisation. An identifier together with its unique mark is called a *Unique name*.

Thus an identifier which occurs in more than one declaration or initialisation corresponds to more than one unique name.

The generation of fresh unique names for identifiers can be suppressed by using the standard routines:

nonunique, unique all \in () \Rightarrow ()

If nonunique is applied, all declarations or initialisations of a given identifier until unique is applied will give rise to the same unique name. This may save storage space and can be used when no confusion is liable to occur.

To sum up:

A new identifier is introduced by introducing a fresh sequence of characters.

A new unique name is introduced by each declaration or initialisation (unless *nonunique* has been applied).

A new variable is introduced by each dynamic activation of a declaration or initialisation.

A variable has an *Extent* which is a sequence of evaluations of expressions and statements.

The extent of a global variable starts from its declaration and continues indefinitely.

The extent of a local or formal variable starts on entry to the body of the function in which it is declared or initialised and continues until exit from the body. During this extent the extent of any other variable with the same unique name is temporarily interrupted. This is called a *Hole in the Extent* of the other variable. Its value is not altered but it cannot be accessed or changed by assignment. Thus there is only one variable *Currently Associated* with a particular unique name during any evaluation. Other variables associated with the unique name are in abeyance.

More than one global declaration of the same identifier is not permitted unless a cancellation of it intervenes in the text.

Similarly a declaration of a local variable is not permitted if there is already a declaration of a local or initialisation of a formal with the same identifier for the same function body.

A Standard Variable is a global variable which already has a value on entry to the POP-2 system. A Standard Function (or Routine) is one which is the value of a standard variable. Certain standard variables are Protected, i.e. no assignment may be made to them.

3.3. Cancellation

A cancellation terminates the scope of any declaration of an identifier and removes the effect of any restrictions placed upon the identifier. The cancellation must occur textually between the old declaration and any new declaration. It may not occur in a function body.

The syntax of cancellations is:

```
\langle cancellation \rangle ::= \mathbf{cancel} \langle identifier * \rangle
```

4. FUNCTIONS

4.1. Definition of functions

A Function is a compound item. Definition and application of functions are treated in this section and the next. Certain properties of a function regarded as a data structure are treated in section 8.7 'Functions'.

A function consists of a Formal Parameter List which is a list of identifiers of formal variables, possibly an output local list which is a list of the identifiers of output local variables (see section 4.2 'Application of functions') and a Body which is an imperative sequence (see section 5.3 'Statements and imperatives').

A function which produces no results (see section 4.2 'Application of functions') is called a *Routine*.

Functions may be referred to in the program by using a function constant, called a *Lambda Expression*, or they may be standard functions provided by the POP-2 system, or they may be created by partial application or by application of a standard function which produces a function as a result.

The syntactic representation of a function constant is:

```
⟨formal parameter list element⟩::=⟨ident⟩ | ⟨restriction⟩
⟨formal parameter list⟩::=⟨formal parameter list element *?⟩
⟨output local list element⟩::=⟨ident⟩ | ⟨restriction⟩
⟨output local list⟩::= ⇒ ⟨output local list element *?⟩
⟨function body⟩::=⟨imperative sequence⟩
⟨lambda expression⟩::=lambda⟨formal parameter list⟩⟨output local list⟩;
⟨function body⟩ end
```

Example:

```
\langle lambda \ expression \rangle e.g. := lambda x \ y ; \ cons(x, \ cons(a, \ y)) end | lambda x ; \ nl(1) ; \ print(x) end
```

We very often wish to declare a variable and then assign a function to it. The syntactic form of this will be as follows:

```
vars \langle identifier \rangle; \langle lambda \rangle \langle formal\ parameter\ list \rangle; \langle function\ body \rangle end \rightarrow \langle identifier \rangle
```

This is so common that a special syntactic form is introduced which is equivalent to it:

```
\( \langle function \rangle ::= \text{function } \rangle routine \\ \langle function \rangle ::= \langle function \rangle \langle formal parameter list \rangle ; \\ \langle function body \rangle \text{end} \\ \end{arrange} \)
```

The word **routine** is a synonym for **function**. It may be used for a function with no results.

If the identifier has been previously declared at this level no new declaration is implied and the function definition is equivalent simply to an assignment of a lambda expression. The identifier may be an operation identifier.

Example:

```
⟨function definition⟩e.g.::= function max \ x \ y; if x > y then x else y close end

| routine enter \ u \ v; cons(conspair(u, v), dict) \rightarrow dict end

| function order \ x \ y \Rightarrow u \ v;

if x > y then x \rightarrow u; y \rightarrow v

else y \rightarrow u; x \rightarrow v close end
```

4.2. Application of functions

An *n-Tuple* is an ordered sequence of n items $(n \ge 0)$. An item is identical with the 1-tuple whose sole member is that item. An n-tuple and an m-tuple may be *Concatenated* to produce an (n+m)-tuple.

A function of n arguments (i.e. with n formal parameters, excluding frozen formals; see section 4.4 'Partial application'), may be Applied to an n-tuple, whose members are called the Actual Parameters of the function. Application of a function to its actual parameters produces an m-tuple, whose members are said to be the Results of the function. A function producing no results (i.e. an 0-tuple) is called a routine (see section 4.1 'Definition of functions').

A function which does not take a fixed number of arguments is called *Variadic*. A function which does not produce a fixed number of results is called *Variresult*.

The application of a function to its actual parameters consists of the following sequence of events:

Entry: a new variable corresponding to each formal parameter is initialised to the corresponding actual parameter value, or if it is a frozen formal to the corresponding value in the frozen value list. A new variable corresponding to each local variable declaration in the function body but not in any interior function body is then created. The variables previously associated with the identifiers of formal or local variables can no longer be referred to but their values are undisturbed.

Running: the function body is evaluated with the variables created on entry.

Exit. Any items which have been placed on the stack (see section 5.3 'Statements and imperatives') and were not there at entry are concatenated with the values of any Output Local Variables to form the results of the function. The variables created on entry are terminated and the variable associated with each identifier reverts to what it was on entry. There is no change in the values of variables which were previously associated with the formal or local variable identifiers and have now been reinstated. The values of formal and frozen variables are lost. The frozen formals will be reinitialised from the frozen value list on the next entry to the function normally with the same values as last time; the frozen value list can be changed by using frozval (see section 8.7 'Functions').

4.3. Nonlocal variables

Variables which occur in a function body and are not locals (i.e. declared in the body) or formals (i.e. elements of the formal parameter list) are called *Nonlocal* to the function. They may be globals or locals of some outer function body which textually encloses it. Care must be taken not to apply a function with nonlocals in a hole in the extent of some of its nonlocals (see section 3.2 'Declaration and initialisation') or outside their extent. Mention of the identifier of such a nonlocal would refer to a quite different variable currently associated with that unique name. The difficulty can arise for recursive functions. Analogous trouble may arise if nonunique is used.

To avoid such difficulties a frozen formal may be used instead of the non-local, provided that it is not desired to assign a new value to the nonlocal as a result of the call. The frozen formal can be initialised by partial application to the value that the non-local would have taken. (Note that the frozen formals can be used in this way to give the equivalent of CPL fixed functions, see CPL Reference Manual privately circulated by C. Strachey, Programming Research Unit, Oxford University.) In cases where assignment to the non-local is desired a frozen formal can be used and initialised to take a reference (see section 8.1 'References') as value. The component of this reference can then be assigned to, and so long as the reference is made the value of some other exterior variable the value is accessible outside the function body.

4.4. Partial application

In section 4.2 'Application of functions' we explained the method of applying a function to its arguments. There is a process somewhat analogous to application called *Partial Application*. By this means some of the formal parameters of a function may be made into *Frozen Formals*, producing a new function with fewer arguments. The frozen formals are always initialised to a fixed value when the function is applied and do not require any corresponding actual parameters (*see* however section 8.7 'Functions' for means of altering this fixed value). In other words the actual parameters corresponding

to the frozen formals are supplied once and for all on partial application. The values of the frozen formals are called the *Frozen Value List*.

For example by partially applying the two argument function 'multiply' to 2 we get a one argument function to double a number, and by partially applying it to 3 we get a function to triple a number. These two functions can coexist, and in general one function can be used to generate any number of others by partial application.

More formally we say that a function f of m arguments may be partially applied to an n-tuple of actual parameters with $n \le m$. We assume for the moment that f has no frozen formals. The partial application produces a new function f' with m-n ordinary formals corresponding to the first m-n formals of f, and n frozen formals corresponding to the last n formals of f. The function f' has a frozen value list consisting of the n items supplied as actual parameters of the partial application.

If f itself has some frozen formals already, say k of them, then f' will have n+k frozen formals and n+k corresponding items in its frozen value list.

The standard function partapply takes a function as its first argument and a list as its second argument, and partially applies the function to the elements of the list.

```
partapply \in function, list \Rightarrow function
```

Note that partial application constructs a new function with a particular frozen value list, it does not alter the original function in any way. A function which has been produced as the result of partial application is called a *Closure Function*. The frozen values of a closure function can be selected or updated (see section 8.7 'Functions').

If a doublet (see section 4.5 'Doublets') is partially applied to one or more items it produces a new doublet. The selector of the new doublet is obtained by partially applying the selector of the original doublet to the given items. The update routine of the new doublet is obtained by partially applying the update routine of the original doublet to the given items.

A special syntactic form is also available for partial application. It is similar to that for ordinary application (see section 5.1 'Expressions').

```
⟨partial application bracket⟩=(%|%)
⟨partial application⟩::=⟨non-operation identifier⟩(%⟨expression list⟩%)
|⟨lambda expression⟩(%⟨expression list⟩ %)
```

The value of the variable currently associated with the identifier is partially applied to the concatenation of the expressions in the expression list. Thus for example:

```
vars c; cons(\%[is \ a \ number]\%) \rightarrow c;

c(1) \Rightarrow

** [1 is a number]

c(2) \Rightarrow

** [2 is a number]
```

```
function f x y z; .. etc. end;
f(\%, y1, z1, \%) \rightarrow f1; f1(x1) \Rightarrow
```

4.5. Doublets

When dealing with data structures, functions called selectors are defined which may be applied to a structure to produce its components (see section 7.1 'Functions of data structures'). To each selector there corresponds an update routine which alters the value of the component in the structure to a given new value.

Any function may have an update routine associated with it. This will normally only be done for selector functions. The function is then called a *Doublet*. When a function is created using a lambda expression its associated update routine is not defined. An update routine may be associated with it by using the doublet *updater*, (see section 8.7 'Functions').

When a variable whose value is a doublet is used as the operator of a compound expression the selector function of the doublet is applied. But when such a variable is used as the operator of a quasi compound expression (i.e. as part of a destination of an assignment) the update routine is applied.

It is convenient to extend our notation for functions (see section 1.5 'Notation for functions') using the new symbol ' $=\Rightarrow$ ' to express concisely the domain and range of the selector and update routines of a doublet. Thus if f is a doublet we write

```
f \in d1, \dots, dk \Longrightarrow r
meaning that f has a selector s
s \in d1 \dots, dk \Longrightarrow r
and an update routine u
u \in r, d1, \dots, dk \Longrightarrow ()
```

Example:

The standard function hd used in list processing (see section 8.3 'Lists') is a doublet.

```
vars l; [1\ 2\ 3\ 4] \rightarrow l; hd(l) \Rightarrow
** 1
5 \rightarrow hd(l); l \Rightarrow
** [5\ 2\ 3\ 4]
hd(l) \Rightarrow
** 5
function second l; hd(tl(l)) end;
lambda x\ l; x \rightarrow hd(tl(l)) end \rightarrow updater(second);
second(l) \Rightarrow
** 2
6 \rightarrow second(l); l \Rightarrow
** [5\ 6\ 3\ 4]
```

4.6. Arithmetic operations

In sections 2.2 'Integers' and 2.3 'Reals' a number of standard functions were introduced for performing arithmetic on integers and reals.

We say that an item is a *Number* if it is either a real or an integer. Arithmetic on numbers is performed by the following standard operations:

Operation	Precedence	Explanation	Result
<	7	less than	truthvalue
>	7	greater than	truthvalue
= <	7	less than or equal	truthvalue
>=	7	greater than or equal	truthvalue
+	5	add	real or integer
. • • <u></u> '	5	subtract	real or integer
*	4	multiply	real or integer
11	4	divide	real
↑	3	exponent	real

These are defined in terms of *intadd*, realadd, etc. and *isreal*, *isint* and realof. +, - and * produce an integer result if both arguments are integer otherwise a real result.

5. EXPRESSIONS AND STATEMENTS

5.1. Expressions

An Expression is either a simple expression, a compound expression, a conditional expression or an imperative expression (see section 5.3 'Statements and imperatives').

A Simple expression is either an identifier or a Constant, a constant being an integer, a real or a structure constant. If the simple expression is an identifier then its value is the value of the variable currently associated with that identifier. If it is a constant then its value is the item denoted by the constant. A Structure Constant is either a lambda expression which is dealt with in section 4.1 'Definition of functions' and in section 8.7 'Functions', a word constant, a string constant or a list constant, all of which are dealt with in section 8 'Standard structures'.

A Compound expression has an Operator which is an expression and some Operands which are an expression list. The value of a compound expression is found by evaluating the operands and evaluating the operator, whose value should be a function (see section 4.5 'Doublets' for the case where the operator is a doublet). The sequence in which these evaluations are carried out is not defined. The function obtained from the operator is then applied to the n-tuple obtained by evaluating the operands. The case where the number of arguments required by the function is not equal to the number of items obtained by evaluating the operands is dealt with in section 5.3 'Statements

and imperatives'. The results of this application are the value of the expression. Thus the value of the expression is an n-tuple, with n=0 if the function is a routine.

Evaluation of conditional expressions is described in section 6.1 'Conditional expressions', and that of imperative expressions in section 5.3 'Statements and imperatives'.

An expression list is evaluated by evaluating the expressions of which it consists and concatenating the results. The order in which the evaluations are made is not defined. The order in which the results of evaluating the expressions are concatenated is the order in which the expressions occur.

The syntax of expressions is given below. There are a number of syntactic forms for compound expressions. A further explanation of the syntax is given in section 5.2 'Precedence'.

```
\langle non-operation identifier \rangle ::= \langle identifier \rangle \mid nonop \langle operation \rangle
       \langle constant \rangle ::= \langle integer \rangle | \langle real \rangle | \langle structure \ constant \rangle
       \langle structure\ constant \rangle ::= \langle lambda\ expression \rangle \mid \langle quoted\ word \rangle \mid \langle string \rangle
                                              ⟨simple expression⟩::=⟨non-operation identifier⟩ | ⟨constant⟩
      \langle operation \rangle ::= \langle identifier \rangle
       \langle parentheses \rangle ::= (|)
       \langle compound\ expression \rangle ::= \langle non-operation\ identifier \rangle (\langle expression\ list \rangle)
                                                |\langle expression ?\rangle operation \langle expression ?\rangle
                                                |⟨closed expression?⟩⟨dot operator*⟩
                                                | \( structure expression \)
       \langle closed\ expression \rangle := \langle simple\ expression \rangle | \langle list\ expression \rangle
                                          | \( conditional expression \)
       \langle dot \ operator \rangle ::= . \langle non-operation \ identifier \rangle
       \langle structure\ expression \rangle := \langle partial\ application \rangle \mid \langle list\ expression \rangle
      ⟨expression list⟩::=⟨expression?⟩, ⟨expression list⟩ | ⟨expression ?⟩
       \langle expression \rangle ::= \langle simple \ expression \rangle \mid \langle compound \ expression \rangle
                                | \( \conditional \) expression \( \) \( \lambda \) imperative expression \( \rangle \)
                                |(\langle expression \ list \rangle)|
Examples:
       \langle simple \ expression \rangle e.g. := x \mid nonop + \mid 3 \mid lambda \ x; x+1 \ end \mid [3.5.9]
       \langle operation \rangle e.g. := + | *** | adjoin
       \langle compound\ expression \rangle e.g. := f(x+1, y) | a*(b+c) | x.hd
                                                     |f(\%x\%)|[\%x,x+1,x+2\%]
       \langle expression \rangle e.g. := a \mid g(h(x+1)) \mid \text{if } x=0 \text{ then } y \text{ else } z \text{ close}
                                      |(x+1\rightarrow x; y+1\rightarrow y; x*y)|
                                     |(x, y+1, z-1)|
```

The various syntactic forms of compound expressions denote the operator and operands in the following way:

(i) $\langle non\text{-}operation identifier \rangle$ ($\langle expression list \rangle$). Here the operator is the identifier and the operands are the expression list.

- (ii) $\langle expression? \rangle$ $\langle operation \rangle$ $\langle expression? \rangle$. This is equivalent to: **nonop** $\langle operation \rangle$ ($\langle expression? \rangle$, $\langle expression? \rangle$) which is a special case of (i) above.
- (iii) (closed expression?). (non-operation identifier). This is equivalent to:
- $\langle non-operation identifier \rangle$ ($\langle closed expression? \rangle$) which is a special case of (i) above.
- (iv) (structure expression). This is equivalent to (i) above with a special identifier for the operand. The exact rules are given in section 4.4 'Partial Application' and section 8.3 'Lists'.

Note that there is no syntactic provision above for compound expressions whose operator is an expression other than an identifier.

5.2. Precedence

If a compound expression or quasi compound expression is of the form

⟨expression ?⟩ ⟨operation⟩ ⟨expression ?⟩

the operator is the operation. In this case ambiguity might arise in the analysis of expressions such as

⟨expression⟩ ⟨operation⟩ ⟨expression⟩ ⟨operation⟩ ⟨expression⟩

which could be analysed with association to the left or to the right. This ambiguity is resolved by the notion of precedence. A precedence is a positive integer between 1 and 7 associated with an operation identifier. It is set by a declaration and can only be changed by cancellation. The operator of a sequence of expressions containing one or more operations is the operation of highest precedence or if there is more than one operation of highest precedence the rightmost of these.

It must be made clear that the difference between an operation and any other identifier which is restricted to having function values is purely a syntactic one.

It may be desired to use an operation in a context other than as the operator of a compound expression. If so it must be prefixed with the word nonop in which case it is treated syntactically like any other identifier. The use of nonop overrules the precedence of the identifier but does not remove restriction of its values to functions. This facility enables operations to appear as operands and enables assignment to operations.

Example:

> has precedence 7, + and - have precedence 5 and * has precedence 4. 5-x+2*y>1+2 is the same as ((5-x)+(2*y))>(1+2).

5.3. Statements and imperatives

A statement is either an assignment, a goto statement, a machine code instruction or an expression list. It may be labelled.

An imperative is either a declaration or a statement. The syntax is:

Example:

```
\langle imperative \ sequence \rangle e.g. ::= loop: x-1 \rightarrow x; f(x) \rightarrow y; \text{ if } x > 0 \text{ then goto } loop; \\ | x+1 \rightarrow y; y; u \rightarrow y; \rightarrow z;
```

The evaluation of an *Imperative Sequence* consists of evaluating the statements in the sequence in which they occur, except when a goto statement occurs and the sequence continues at the point indicated by the goto statement.

An Imperative Expression may be formed from an imperative sequence. The syntax is:

```
\langle imperative \ expression \rangle := (\langle imperative \ sequence \rangle)
```

The Stack is an ordered sequence of items. The last item to be added to this sequence is said to be on Top of the Stack. Items can be added to the top of the stack or removed from the top of the stack. On entry to the POP-2 system the stack is empty. When a statement is evaluated any results produced are added to the top of the stack. The results of an imperative sequence are the items left on the stack when the sequence has been evaluated.

Evaluation of a statement which is a compound expression may affect the stack as follows. If the number of arguments required by the function obtained by evaluating the operator is not the same as the number of items produced by evaluating the operands, these items are loaded on to the stack in sequence. The function then takes its arguments off the stack, the last argument being the one which was on the top of the stack. Thus suppose that the function requires m arguments and the operands yield n items. If m > n the first m - n arguments are taken off the stack. If m < n the first n - m items produced by the operands are left on the stack. If m = n the stack is not affected by evaluating the compound expression. Exactly analogous remarks apply to quasi compound expressions.

5.4. Labels and goto statements

A Label may be attached to a statement. Evaluation of a Goto Statement using that label causes the sequence of evaluation to be changed so that the labelled statement is evaluated next. A goto statement may not refer to a label outside the function body in which it occurs. If a goto statement occurs

in an operand of a compound or quasi compound expression it may not refer to a label outside that operand. The syntax is:

```
\langle labelled statement \rangle ::=\langle label \rangle :\langle statement ?\rangle \langle goto \statement \rangle :=\langle identifier \rangle \langle label \rangle ::=\langle identifier \rangle \rangle identifier \rangle identifier
```

The statement return causes transfer of control to the exit of the innermost current function body. There is a standard macro exit which is synonymous with return close.

If an identifier or sign is used for a label it may not appear as an identifier associated with a variable in the text constituting that function body.

Goto statements and labelled statements may only occur inside a function body.

Note that a label is not an item.

Example:

```
loop: x+1 \rightarrow x; y*y \rightarrow y;
if x=0 then goto loop close.
```

5.5. Assignment

An Assignment consists of a Source, which is an expression or sequence of expressions and a Destination List, which is a sequence of elements each of which is either an identifier or a Quasi Compound Expression.

A quasi compound expression has an operator which is an expression and some operands, i.e. a sequence of expressions (possibly an empty sequence). Note that a quasi compound expression is not an expression and cannot be evaluated alone to produce an item; it is merely a component of an assignment. It is syntactically the same as a compound expression but cannot be evaluated in isolation.

An assignment is evaluated as follows. First the source is evaluated to yield an n-tuple (where $n \ge k$, k being the number of destination elements). The last k elements of this n-tuple, which we will call the *Source Items*, are then taken in sequence starting from the last and each source item is combined with the corresponding destination element (taken in sequence starting from the first) as follows:

- (i) If the destination element is a variable the source item becomes the new value of that variable.
- (ii) If the destination element is a quasi compound expression the operator and operands of this expression are evaluated. The value of the operator must be a doublet (see section 4.5 'Doublets') and its update routine is applied to the concatenation of the source item and the values of the operands.

The syntax of quasi compound expression is given below. A further explanation of this syntax is given in section 5.2 'Precedence'.

\(\langle quasi \) compound \(expression \rangle ::= \langle non-operation \) identifier \(\langle \) (\(expression \)

```
|\langle expression ?\rangle \langle operation \rangle \langle expression ?\rangle \langle dot operator * ?\rangle
```

The syntax of assignments is:

```
\langle assignment \rangle ::= \langle expression \ list \rangle \langle destination * \rangle \ | \langle function \ definition \rangle 
 | \langle macro \ definition \rangle 
 \langle destination \rangle ::= \rightarrow \langle non-operation \ identifier \rangle \ | \rightarrow \langle quasi \ compound \ expression \rangle
```

Example:

```
\langle assignment \rangle e.g. := x+1 \rightarrow y \mid u+v \rightarrow a(i,j) \mid x//y \rightarrow u \rightarrow v
```

In the second example a(i, j) is a quasi compound expression and the whole assignment is a euphemism for a1(u+v, i, j) where a1 is the update routine of the doublet a.

Function definitions and macro definitions are special syntactic forms for assignments.

5.6. Comments

The word **comment** and all characters after it up to and including the next semicolon are ignored.

6. CONDITIONALS

6.1. Conditional expressions

A conditional expression is composed of three components which we will call the *Condition*, the *Consequent* and the *Alternative*. The condition is an expression with a single result, a conjunction or a disjunction (see section 6.2 'Conjunctions and disjunctions'). The consequent and the alternative are imperative sequences each having the same number of results. The method of evaluation of a conditional expression is as follows:

The condition is first evaluated. If its value is the truth value *true* then the consequent is evaluated and its value becomes the value of the expression. But if the value of the condition is the truth value *false* then the alternative is evaluated and its value becomes the value of the expression.

The alternative of a conditional expression may be omitted if it is an empty imperative sequence.

It will often happen that the alternative is itself a conditional expression. The syntax of conditionals is arranged to provide a compact notation to express this:

```
⟨elseif clause⟩::=elseif ⟨condition⟩ then ⟨conditional body⟩
⟨else clause⟩::=else ⟨conditional body⟩
```

Example:

```
\langle conditional\ expression \rangle e.g. := if\ x>0 and x<3 then y else if x>3 then z else 0 close | if x=0 then 1\rightarrow y close
```

If there are no elseif clauses the conditional body is the consequent and the else clause is the alternative (which may be omitted). If there are elseif clauses then the first expression is the condition, the second is the consequent and the remainder is the alternative, and it is to be regarded as the conditional expression obtained by replacing the first elseif by if and inserting an extra close before the close, e.g.

if p then x else if q then y else z close is equivalent to

if p then x elseif q then y else z close close.

6.2. Conjunctions and disjunctions

A Conjunction is composed of two component expressions each producing a single result. The method of evaluating a conjunction is to evaluate the first component expression and if its value has the truth value false the value of the conjunction has truth value false, otherwise the second expression is evaluated and the conjunction has a truth volue equal to that of the second component expression.

A Disjunction is composed of two component expressions each producing a single result. The method of evaluating a disjunction is to evaluate the first component expression and if its value has the truth value true the value of the disjunction has truth value true, otherwise the second expression is evaluated and the disjunction has a truth value equal to that of the second component expression.

A number of conjunctions and disjunctions can be combined to form a condition.

The syntax is:

```
\( \condition \) ::=\( \center \)::=\( \center \) and \( \condition \) | \( \center \) | \( \center \) condition \( \center \)
```

These three kinds of conditions are respectively a conjunction, a disjunction and an expression.

Thus and and or associate to the right.

$$\langle condition \rangle e.g. := x < 10 \text{ and } x > 0 \mid x > 10 \text{ or } x < 0 \mid null(x) \mid b \mid p(x) \text{ and } q(x) \text{ or } r(x)$$

In the last example the following cases can occur ('-' means that the expression is not evaluated)

p(x)	q(x)	r(x)	value of condition
false			false
true	false	false	false
true	false	true	true
true	true	. :	true

7. DATA STRUCTURES

7.1. Functions of data structures

A Data Structure is a compound item which has other items as its Components. For each class of data structures there is a family of functions called the Characteristic Functions acting upon structures of that class. These functions are a constructor, a destructor, selectors and update routines. A given compound item may represent a number of different data structures by being used in association with more than one family of functions and hence having different components.

Given values for its components it is possible to construct a data structure using a *Constructor* function, say c.

 $c \in component, ..., component \Rightarrow data structure$

It is possible to select the value of a component of a data structure. For each component there is a *Selector* function, say si.

 $si \in data \ structure \Rightarrow component$

It is possible to update a component of a data structure, i.e. to give it a new value. For each component there is an *Update Routine*, say *ui*:

 $ui \in component, data structure \Rightarrow ()$

When a data structure is updated the old version is overwritten.

It is convenient to define another function called a *Destructor* function, say d, which is the inverse of the constructor, i.e. given a data structure it produces its components as results.

 $d \in data \ structure \Rightarrow component, \dots, component$

After applying the destructor to a structure, the structure is deleted.

There is a relation called equality (see section 2.1 'Simple and compound items') which may hold between two compound items. It is denoted by the standard function = (an operation of precedence 7). This function is also defined for simple items with the usual meaning.

 $= \in item, item \Rightarrow truthvalue$

Thus if the value of an expression E1 is equal to the value of an expression E2 then the expression

E1 = E2

has value true.

Equality means that the two compound items contain the same address, i.e. they point to the same area of store. If the items are not equal they point to entirely different areas of store. We say that a compound item is *Copied at the Top Level* if a new item is formed pointing to a new area of store which contains items equal to those of the given compound item. The new item and the previous one are not equal. They are however *Equivalent*.

Equivalent compound items are defined as items which are either equal or all of whose components are equivalent.

Updating an item alters a component item in the store area pointed to by the item but does not cause copying.

We will now give a more formal explanation of equality, but the model in terms of addresses and storage may be kept in mind.

Equality is an equivalence relation, i.e. it is

- (i) reflexive (x=x);
- (ii) symmetric (if x=y then y=x); and
- (iii) transitive (if x=y and y=z then x=z).

It has the following other properties:

- (iv) The value of a formal parameter variable is equal to the corresponding actual parameter.
- (v) If an item is assigned to a variable then the value of the variable is equal to that item.
- (vi) An item, other than a word or simple item, which is read in (see section 9.1 'Input') is not equal to any other item.
 - (vii) The rules for equality of words are given in section 8.6 'Words'.
- (viii) Two integers or two reals are equal according to the usual rules of arithmetic. An integer is never equal to a real.

Items are equal only if their equality follows from the above properties.

We can now state some relationships between the various functions on data structures. We will use a and b for data structures, $x1, \ldots, xi, \ldots, xk$ for items occurring as components, $s1, \ldots, si, \ldots, sk$ for selectors, $u1, \ldots, ui, \ldots, uk$ for update routines, c for a constructor and d for a destructor.

- (i) $s1(a), \ldots, sk(a)$ is the same *n*-tuple as d(a), i.e. they have equal elements.
 - (ii) si(c(x1,...,xi,...,xk)) = xi is true.
- (iii) c(s1(x), ..., sk(x)) = x is always false, but the left-hand expression is equivalent to x.
 - (iv) After evaluating ui(xi, a),

$$si(a) = xi$$
 is true.

- (v) After evaluating ui(si(a), a), a is unchanged.
- (vi) If a=b then ui(xi, a) is evaluated, si(b)=xi is true and a=b is still true.
- (vii) From (i) and (ii) above d(c(x1,...,xk)) is the same n-tuple as x1,...,xk, i.e. they have equal elements.

If a and b are data structures and a is not equal to b and updating a com-

ponent of a also updates some component of b then a and b are said to Share.

When we wish to discuss a class of data structures which do not all have the same number of components (such as strips, see section 7.3 'Strips') it is convenient to define a General Selector function and a General Update Routine.

The general selector function, say s, has as arguments, an integer, i, and a data structure. It selects the ith component of the data structure.

 $s \in integer$, data structure \Rightarrow component

Thus if si is the ith selector s(i, a) = si(a). Similarly for the general update routine, say u,

 $u \in component, integer, data structure \Rightarrow ()$

Thus if ui is the *i*th update routine, u(xi, i, a) has the same effect as ui(xi, a).

The programmer is able to create new kinds of data structures called records and strips (see section 7.2 'Records' and 7.3 'Strips'). He can also create functions by methods already described. He may be able to create other kinds of data structures using extra standard functions or machine code but this depends on the implementation. Certain classes of records and strips are standard and these are described in section 8 'Standard structures'.

There are a number of special expressions called 'structure expressions' used to construct these standard structures (see section 5.1 'Expressions').

Given a class or several classes of data structures with their associated functions it is possible to define functions which characterise a new family of data structures. Suppose for example that we have a class of structures with two selectors, say s1 and s2, and components which are full items and members of the same class of structures. We can then define a new class of structures whose selectors are given by:

```
function s11\ a; s1(s1(a)) end; function s12\ a; s1(s2(a)) end; function s21\ a; s2(s1(a)) end; function s22\ a; s2(s2(a)) end;
```

If c is the constructor of the first class we define the new constructor:

```
function cc \ x1 \ x2 \ x3 \ x4; c(c(x1, x2), c(x3, x4)) end;
```

Note that if it is associated with two or more families of functions the same compound item can represent two or more structures, one of each class. However, for each class of compound items there is one *Primitive Data Structure Class* and other data structures are defined in terms of this primitive class. A primitive data structure does not share with any other primitive data structure.

7.2. Records

A Record is a compound item which is a member of a Record Class. The Size of a set of items is an integer item. If all the items in the set are restricted

to be non-negative integers less than 2^n , the size is the integer n, otherwise if the component is a *Full Item* (i.e. the set is not restricted) the size is the integer 0. For each component of a record there is a size associated with the set of possible values of that component. The *Specification of a Record* is the list of sizes associated with its components. A record class is a set of records which all have the same specification, and this is said to be the specification of the record class. Note that a record class is not an item. A word is associated with each record class.

A family of functions is associated with each record class to form a primitive class of data structures. This family comprises a set of selectors (\in record \Rightarrow component) and a set of corresponding update routines (\in component, record \Rightarrow ()), a constructor (\in component, ..., component \Rightarrow record) and a destructor (\in record \Rightarrow component, ..., component). Each selector function may be paired with the corresponding update routine to form a doublet (\in record $=\Rightarrow$ component). The standard function recordfns is used to create a new record class. It requires as arguments the word to be associated with the record class, an estimate of the number of records in the record class (this is purely to help in efficient implementation) and the specification of the record class. It produces the constructor, the destructor and the doublets for the record class. The number of its results depends on the length of the specification list. Normally the programmer will immediately assign these resulting functions to variables.

 $recordfns \in word$, integer, $specification \Rightarrow constructor$, destructor, doublet, ..., doublet

There is a standard function which converts a record to a list of its components:

 $datalist \in record \Rightarrow list$

There is a function *dataword* which given a record produces the word associated with its record class.

 $dataword \in record \Rightarrow word$

The function *copy* copies a record at the top level.

 $copv \in record \Rightarrow record$

The functions *datalist*, *dataword* and *copy* are defined over records of any class, and whenever *recordfns* is used to create a new record class these three functions are extended to deal with records of that class.

The routine *enddata* may be given the word associated with a record class and removes all records in that class. It also adjusts the three functions just mentioned so that they no longer deal with that record class.

 $enddata \in word \Rightarrow ()$

7.3. Strips

A Strip is a compound item which is a member of a Strip Class. All components of a strip must have the same size (see section 7.2 'Records'

for definition of size) which is called the *Component Size* of the strip. All strips in a strip class must have the same component size but not necessarily the same number of components. A word is associated with each strip class.

A family of functions is associated with each strip class to form a primitive class of data structures. This family includes a general selector function $(\in integer, strip \Rightarrow component)$ and a general update routine $(\in component, integer, strip \Rightarrow ())$. The selector function may be paired with the update routine to form a doublet. It also includes for each strip class an initiator function $(\in integer \Rightarrow strip)$. This constructs a strip with the given number of components, but the values of these components are not defined. The initiator may be used with the update function to define a constructor function for strips of the strip class.

The standard function *stripfns* is used to create a new strip class. It takes as arguments the word to be associated with the strip class, an estimate of the total number of all components of all strips in the strip class (this is purely to help in efficient implementation) and the component size of the strip class. It produces as results the initiator function and the doublet for the strip class:

 $stripfns \in word$, integer, $size \Rightarrow initiator$, doublet

There is a standard function which converts a strip to a list of its components. This is *datalist* (see section 7.2 'Records'). There is a function which given a strip produces the word associated with its strip class. This is *dataword* (see section 7.2 'Records').

The function *copy* copies a strip at the top level (see section 7.2 'Records').

The functions datalist, dataword and copy and the routine enddata act for strips just as for records.

7.4. Garbage collection

Storage for the construction of data structures is made available by a storage control system. This system must be able to make use of areas of store which have been used but are no longer required. This is achieved by a process known as *Garbage Collection* which is undertaken whenever the system runs short of store. This first of all discovers what items can still be referred to by the programmer, e.g. because they are the value of a variable whose extent has not finished (*see* section 3.2 'Declaration and initialisation'). Any items which can no longer be referred to are destroyed, i.e. their storage area is returned to the system for use in constructing other items. Since he cannot refer to them the programmer is not aware of this destruction.

If variables refer to compound items which are no longer in use, the garbage collector cannot recover the associated storage. The variable should be reset, e.g. to zero. In the case of identifiers the identifier can be cancelled (see section 3.3 'Cancellation').

To avoid too frequent garbage collection compound items can be deleted, i.e. returned to the storage control system, using the standard routine:

 $delitem \in item \Rightarrow ()$

When an item is deleted its components are not deleted.

After an item has been deleted it is no longer available and the onus is on the programmer not to use it. The implementation may not give an error message if he does use it, the value simply not being defined.

8. STANDARD STRUCTURES

8.1. References

There is a standard record class called *References*. These have one component which is a full item. The word associated with the class is "ref". Thus before entry to the POP-2 system this class is created using recordfns, and the resulting functions are assigned to variables to give the following standard functions:

constructor: $consref \in item \Rightarrow reference$ destructor: $destref \in reference \Rightarrow item$ doublet: $cont \in reference \Rightarrow item$

A reference may be used e.g. as an actual parameter of a function to enable the function to cause side effects by updating the reference.

8.2. Pairs

There is a standard record class called *Pairs*. Records of this class have two components which are both full items. The word associated with the class is "pair". Thus before entry to the POP-2 system this class is created using recordfns, and the resulting functions are assigned to variables to give the following standard functions:

constructor: $conspair \in item$, $item \Rightarrow pair$ destructor: $destpair \in pair \Rightarrow item$, item doublets: front, back $all \in pair = \Rightarrow item$

An Atom is an item which is not a pair. Atoms are recognised by the standard function atom.

 $atom \in item \Rightarrow truthvalue$

8.3. Lists

There is a standard data structure called a *Link* which is used to construct another data structure called a *List*. Lists in POP-2 include structures analogous to LISP lists, but also structures which compute the elements dynamically (cf. P. J. Landin's 'streams').

The word "nil" is used to represent the Null List and the standard variable

nil takes this value on entry to the POP-2 system. The null list is also represented by a pair whose front is *true* and whose back is a function.

The standard function null recognises the null list

```
null \in list \Rightarrow truthvalue
```

A list is either the null list or it is a link.

A link is either:

- (i) a pair whose front component is any item and whose back component is a list, or
- (ii) a pair whose front component is false and whose back component is a function with no arguments and one result.

In case (ii) the function is one which when repeatedly applied produces a succession of items, not necessarily all the same, i.e. normally the function will side-effect itself. The last item produced should be the terminator. For example this enables us to convert an input file to a list. Lists with this sort of link are dynamic and some or all of their elements are computed rather than stored statically.

The characteristic functions of a link are:

```
constructor: cons \in item, list \Rightarrow link

destructor: dest \in list \Rightarrow item, list

doublets: hd \in link = \Rightarrow item (called the 'head')

tl \in link = \Rightarrow item (called the 'tail')
```

These functions are very similar to those for pairs, but in the case of a link of the second kind special precautions are taken to make sure that on applying the selector tl the front component is not lost but preserved in a pair. Thus if x has a list as its value and tl(x) is evaluated there is a side effect on x, but matters are so arranged that this side effect is not detectable using the list processing functions. The function cons is the same as conspair and produces a link of the first kind. The following standard function produces a link of the second kind or the null list given a function of no arguments:

```
fntolist \in (() \Rightarrow item) \Rightarrow list
function fntolist \ f; \ cons(false, f) \ end
```

The other characteristic functions are defined as follows:

First an auxiliary function (not standard) to convert the first link of a dynamic list to static form.

```
function solidified 1; vars f x;

if isfunc(back(1))

then back(1) \rightarrow f; f() \rightarrow x;

if x = termin then true \rightarrow front(1)

else x \rightarrow front(l); conspair(false, f) \rightarrow back(l)

close; l

else 1

close

end;
```

```
function hd l; front(solidified(l)) end;
lambda i \ l; \ i \rightarrow front(solidified(l)) \ end \rightarrow updater(hd);
function tl\ l; back(solidified(l)) end;
lambda i \ l; \ i \rightarrow back(solidified(l)) end\rightarrow updater(tl); \cdot
function dest l; vars f;
  if isfunc(back(l)) then back(l) \rightarrow f; (f(),l)
     else (front(l),back(l))
   close
end:
function null 1;
  if l=nil then true
      elseif isfunc(back(l))
         then if hd(l) or null(solidified(l)) then true else false close
      else false
   close
end;
```

A list may have no components (if it is the null list) or one or more (if it is a link).

If it is a link its first component is the head component of the link and its remaining components are the components of the list which is the tail component of the link. Thus the characteristic functions for lists can be defined in terms of those for links.

There are two special syntactic forms for constructing lists. These are list constants and list expressions. List constants may have lists, integers, reals, words or strings (see section 8.4 'Full strips and character strips') as components. The list is constructed at compile time.

```
\langle list constant brackets\rangle ::=[|]
\langle list constant \rangle ::=[\langle list constant element *?\rangle]
\langle list constant element \rangle ::=\langle list constant \rangle | \langle character group \rangle
```

Example:

```
\langle list\ constant \rangle e.g. := [1\ 2\ DOG\ CAT] \mid [\ [\ 1\ 2\ ]\ [4\ 5]\ 6]
```

List expressions are formed by evaluating a number of expressions at run time and constructing a list.

```
⟨list expression brackets⟩::=[%|%| %| 
⟨list expression⟩::=[%⟨expression list⟩%]

Thus
[% %] is equivalent to nil
and [%⟨expression⟩ %] is equivalent to cons(⟨expression⟩, nil)
and [%⟨expression⟩, ⟨expression list⟩ %] is equivalent to
cons(⟨expression⟩, [%⟨expression list⟩ %])

Example:
⟨list expression⟩e.g.::=[%x+1, [%x+2, x+3%], tl(y)%]
```

For convenience the following functions are standard:

 $next \in list \Rightarrow item$, list (similar to dest but non-destructive)

::(a synonym for cons but an operation of precedence 2)

 $\langle \rangle \in list, list \Rightarrow list$ (concatenates the lists, an operation of precedence 2).

8.4. Full strips and character strips

Two strip classes are standard.

The first is *Full Strips* with full items as components and associated word "strip". The characteristic functions are:

initiator: $init \in integer \rightarrow full \ strip$ doublet: $subscr \in integer$, $strip = \Rightarrow item$

The second is *Character Strips* (also called 'Strings') (for characters see section 8.6 'Words') with component size 6 and associated word "cstrip". The characteristic functions are:

initiator: $initc \in integer \Rightarrow character\ strip$ doublet: $subscrc \in integer$, $strip = \Rightarrow integer\ of\ size\ 6$

The components of a character strip may be any integers of size not more than 6, they need not necessarily be used to represent characters.

There is a structure constant to construct character strip constants at compile time.

```
⟨string bracket⟩::=/|\
⟨string constant⟩::=/⟨string constant element * ?⟩\
⟨string constant element⟩::=⟨string constant⟩ | ⟨any character except
a string bracket⟩
```

Example:

```
\langle string\ constant \rangle e.g. := \langle ...\ rubbish\ .\ please\ type\ \langle sorry \rangle
```

Spaces and newlines are significant in string constants.

There are functions to input and output character strings stored as character strips (see section 9.1 'Input and output'). The external format is as for string constants.

8.5. Arrays

Arrays give a convenient method of accessing and updating structures indexed by integers. An array has components, which are items of a given size. Each component is associated with a sequence of integers called *Subscripts*. The number of subscripts is known as the number of *Dimensions* of the array. An array is a doublet:

```
array \in subscript, ..., subscript \Longrightarrow component
```

This is in contrast to strips which have a general selector and a general update routine associated with a whole class of strips and take the actual strip referred to as a parameter. Arrays can be formed from strips (or from other data structures) by using partial application. The programmer is free

to do this in any way he chooses but standard functions for creating arrays are provided.

There is a standard function to create a many dimensional array of items of any size. Updating a component of this array does not affect any other component. This function is:

```
newanyarray \in boundslist, (subscript, ..., subscript \Rightarrow component), strip initiator, strip doublet \Rightarrow array
```

The array produced will normally be immediately assigned to a variable. The boundslist is a list of integers, these two integers being alternately the

lower and upper bounds for each subscript. The second parameter is a function used to initialise the components of the array. It must produce the appropriate component for each combination of subscripts. The strip doublet and strip initiator are the characteristic functions of a strip class whose components are of the same size as that required for the array components.

There is also a standard function to create arrays of full items:

```
    newarray ∈ boundslist, (subscript, ..., subscript ⇒ component) ⇒ array
    This is obtained by partial application and is equivalent to
    newanyarray (% init, subscr %).
```

8.6. Words

There is a standard record class called *Words*. It has 8 components of size 6 called *Characters*, and a component called the *Meaning*. The word associated with the record class is "word". The standard functions characterising words are:

```
constructor: consword \in character, ..., character, integer \Rightarrow word destructor: destword \in word \Rightarrow character, ..., character, integer, item doublets: charword \in word = \Rightarrow character, ..., character, integer meaning \in word = \Rightarrow item
```

Each character of the POP-2 character set corresponds to a unique integer. The correspondence rule depends on the implementation. Note that the functions above are variadic and work on a variable number of characters followed by that number as an integer. If there are less than 8 characters supplied to the constructor the remaining character components are not defined and they are not produced by the destructor or selector. The constructor does not take a meaning component as argument. The meaning of a word is *undefined* unless the word has been updated to have a particular meaning.

Words may occur in the program as quoted words, i.e. word constants, with the following syntax:

```
\( \text{unquoted word} \cdots := \left\{ alphanumeric * ?\} \left\ \left\{ sign *\} \\ \left\ \left\{ decorated bracket} \right\ \left\{ bracket decorator} \\ \left\ \left\{ separator} \right\ \left\{ period} \right\ \left\ \left\{ quote} \\ \end{array}
```

\(\decorated \text{ bracket} \) ::=(|) | (\% | \%) | [|] | [\% | \%]

 $\langle quoted\ word \rangle ::= "\langle unquoted\ word \rangle"$

Example:

 $\langle quoted\ word \rangle e.g. := "big" \mid "++" \mid " \%)" \mid """$

Words may also occur as components of constant lists (see section 8.3 'Lists'). Only the first 8 characters are significant.

Words may also be read as data (see section 9.1 'Input'). Words which occur as constants or are read as data are Standardised, i.e. if a word with the same characters already exists no new word is constructed and the compound item produced is the previously existing word, but if no such word exists a new word is constructed with undef as its meaning. Words constructed using consword are also standardised, but the update routines do not standardise.

8.7. Functions

Functions are compound items. There is no constructor or destructor for functions. They can be constructed by the methods described in section 4.1 'Definition of functions', and they can be deleted by the routine *delitem* (see section 7.4 'Garbage collection'). There is a family of characteristic functions associated with the class of functions to form a primitive class of data structures.

Functions have an accessible component which may be used to associate extra information with the function. It is accessed by the standard doublet

$$fnprops \in function = \Rightarrow item$$

Functions have an update routine (see section 4.5 'Doublets'). For a function constructed by using lambda or function or routine this has initially no defined value. This component may be selected or updated by using a standard doublet:

```
updater \in function = \Rightarrow routine
```

Closure functions i.e. those constructed by partial application, have a doublet to select or update the values of their frozen formals

```
frozval \in integer, closure\ function = \Rightarrow item
```

The integer determines which of the frozen formal values is affected, counting from the front (if a closure function is obtained by successive partial applications only the formals frozen by the last one are counted). There is also a doublet to select the function from which the closure function was constructed or replace it with another function.

```
fnpart \in closure function = \Rightarrow function
```

The standard function = follows the usual rules for compound items when applied to functions, i.e. equality is preserved over assignment, updating and actual parameter/formal parameter correspondence but each construction of a function produces a different one.

The following standard function recognises functions:

 $isfunc \in item \Rightarrow truthvalue$

9. INPUT AND OUTPUT

9.1. Input

Information which is input to the POP-2 system is organised into *Files*, each of which comes from a *Device*.

Before a file can be accessed it must be *Opened*. From then on it can be read one character at a time. Eventually it must be *Closed*.

The naming of files and devices depends on the operating system of the implementation. The names of files are lists and the names of devices may be any item. A device name may refer to more than one device.

There is a standard variresult function *popmess* used for communicating with the operating system

```
popmess \in list \Rightarrow item, ..., item
```

This is used for various input and output purposes.

To open a file from a given device, *popmess* is used to produce a function to read characters from it i.e. a function \in () \Rightarrow character. The list supplied to *popmess* has a head which is an input device name and a tail which is a file name.

To close a file before reaching the end of it, *popmess* is again used. The list supplied to it has a head which is the word "close" and a tail which is a list of one element: a character reading function obtained when the file was opened. No result is produced by *popmess* in this case.

The sequence of characters making up a POP-2 text may be split up into Character Groups each of which represents a Text Item. A text item is either an integer, a real, a word or a string. It is represented by a character group, thus:

```
\langle character\ group \rangle ::= \langle integer \rangle \mid \langle real \rangle \mid \langle unquoted\ word \rangle
\mid \langle string\ constant \rangle
```

Character groups are terminated by spaces or newlines where necessary to separate them from the following character group.

There is a standard function to convert a function which produces a character whenever it is applied into a corresponding one which produces a text item whenever it is applied.

$$incharitem \in (() \Rightarrow character) \Rightarrow ((l) text item)$$

The program is input on a standard file called the *Standard Input File* from a standard device called the *Standard Input Device*. There is a standard function to read characterx from the *standard input* file:

$$charin \in () \Rightarrow character$$

The program is compiled from the text item list which is the value of the standard variable *proglist*. Initially this has as value the list of text items from the standard input file. It may be assigned to by the programmer who wishes to compile from a different source.

For convenience there is a standard function *itemread* producing the next item of the list which is the value of *proglist*.

```
itemread \in () \Rightarrow text item
```

It is defined thus:-

function itemread; proglist . dest \rightarrow proglist end;

9.2. Output

Information which is output from the POP-2 system is organised into files, each of which is sent to a device (see section 9.1 'Input').

To open an output device the standard function popmess (see Section 9.1 'Input') is used to produce a routine to deliver characters to it i.e. a routine \in character \Rightarrow (). The list supplied to popmess has a head which is an output device name and a tail which is a file name.

There is a standard function to convert a routine which delivers a sequence of characters to an output file into one which delivers a sequence of text items.

```
outcharitem \in (character \Rightarrow ()) \Rightarrow (text\ item \Rightarrow ())
```

Compiler messages and results of computation are normally output on a standard file called the *Standard Output File* to a standard device called the *Standard Output Device*. There is a standard routine to output characters to the standard output file:

```
charout \in character \Rightarrow ()
```

There is a standard variable *cucharout* which contains the routine to output characters to the *Current Output File*. This contains initially the routine for the standard output file but it may be assigned to if a different output file is to be made current. An output file is closed by outputing the terminator.

There are standard routines to output spaces or newlines to the current output file:

```
sp \in integer \Rightarrow ()
nl \in integer \Rightarrow ()
```

There is a standard function which outputs any item to this file in some suitable format and produces that item unchanged as its result.

```
print \in item \Rightarrow item
```

There is a standard macro which uses *print* and causes the items on the stack starting at the bottom to be printed on a newline preceded by two asterisks. These items are removed from the stack. In a function body only the top item of the stack is affected. This macro is denoted by the POP-2

identifier \Rightarrow (not to be confused with the \Rightarrow used in this manual to show the type of functions). A semicolon is implied before and after so that immediate evaluation can occur (see section 11.1 'Immediate evaluation').

MACHINE CODE

It is possible to insert sections of machine code in an imperative sequence. The rules depend on the implementation. A code instruction is represented by the identifier \$ followed by any sequence of characters which do not include ':'

 $\langle code\ instruction \rangle ::= \$ \langle any\ sequence\ of\ characters\ other\ than\ ; \rangle$

11. MODES OF EVALUATION

11.1. Immediate evaluation

A POP-2 program consists of a sequence of imperatives and cancellations:

```
⟨program element⟩::=⟨imperative⟩ | ⟨cancellation⟩
⟨program⟩::=⟨program elementt⟩;⟨program⟩
```

The program elements are evaluated in sequence in the same way as an imperative sequence. Each program element is evaluated as soon as the terminal semicolon and a space has been read by the compiler. The body of any function in the program element will be compiled and kept so that it may be evaluated when that function is applied.

11.2. Macros

A Macro is a routine which is applied at compile time.

The definition of a macro routine is similar to that of any other routine except that macro is used instead of routine and no formal parameters are allowed.

```
\langle macro\ definition \rangle ::= macro\ \langle identifier \rangle; \langle function\ body \rangle end
```

A macro, like an operation, is applied whenever it is mentioned and does not need parentheses after it.

Although a macro has no parameters the function *itemread* (see section 9.1 'Input') may be used to read the text items following the macro identifier. There is a standard routine which, when applied in a macro body to a list of text items, concatenates these items to the right of the macro identifier in the program sequence of text items.

```
macresults \in text \ item \ list \Rightarrow ()
```

If it is applied more than once it concatenates to the right of the previously inserted items. On exit from the macro the inserted text items are evaluated as program.

Example:

```
macro -\rightarrow; vars x y; itemread \rightarrow x; itemread \rightarrow y; macresults ([% "\rightarrow", y, "\rightarrow", x%]) end; 7/\!\!/2 - \rightarrow q r; This is the same as 7/\!\!/2 \rightarrow r \rightarrow q;
```

The correspondence between a list of text items and POP-2 program is as follows. Syntax words, identifiers and unquoted words are represented by corresponding words in the list. A quoted word is represented by a word with the word quote (consisting of the character quote) before and after it in the list. Integers and reals are represented by integers and reals. String constants are represented by strings.

11.3. Evaluation of program text

A standard function *popval* is provided which will evaluate a list of text items treating it as a POP-2 imperative sequence. The sequence is evaluated immediately. It may contain function definitions and assignments to current variables. Any declarations in it which are not in a function body are global. The list must terminate with the word **goon**. For the correspondence between a list of text items and POP-2 program *see* section 11.2 'Macros'.

The result of the application of *popval* is the result of evaluating the imperative sequence.

```
popval \in text \ item \ list \Rightarrow item, \dots, item
```

Note that *popval* is used to evaluate an imperative sequence at run time and the list of text items may have been produced as the result of computation. It may temporarily affect the standard variable *proglist* (see section 9.1 'Input'). Example:

```
1 \rightarrow a; popval([vars x; a+2 \rightarrow x; x*x goon]) \Rightarrow
```

The standard routine *setpop* may be applied in the imperative sequence. This restores the system to execute mode. The stack is cleared. The variable currently associated with any identifier is not altered. After *setpop* has been applied the rest of the imperative sequence is ignored and all function bodies currently being evaluated are abandoned. The system then evaluates the next program element. *setpop* may also be applied in a function body.

$$setpop \in () \Rightarrow ()$$

ACKNOWLEDGMENTS

This language is a development of R. J. Popplestone's 'POP-1' programming language (see the paper in this volume). The debt to the ALGOL, LISP, CPL and

ISWIM programming languages should be obvious. We are indebted to a number of people in this department and elsewhere for helpful discussion and criticism, to Dr David Park who contributed to discussion of the storage control scheme and to Mrs Margaret Pithie and Miss Eleanor Kerse who typed this report. Mr M. Healy kindly pointed out a number of errors and obscurities in a draft.

The work has been undertaken on a grant from the Science Research Council under the supervision of Dr D. Michie, whose encouragement has been invaluable.

INDICES

TECHNICAL TERMS

actual parameters: 4.2 alternative: 6.1 applied: 4.2 assignment: 5.5 atom: 8.2 body: 4.1 character groups: 9.1 character strips: 8.4 characteristic functions: 7.1 characters: 8.6 closed: 9.1 component size: 7.3 components: 7.1 compounds: 2.1 compound expression: 5.1 concatenated: 4.2 condition: 6.1 conjunction: 6.2 consequent: 6.1 constant: 5.1 constructor: 7.1 copied at the top level: 7.1 currently associated: 3.1, 3.2 data structure: 7.1 declaration: 3.2 destination list: 5.5 destructor: 7.1 device: 9.1 dimensions: 8.5 disjunction: 6.2 doublet: 4.5 entry: 4.2 equivalent: 7.1 exit: 4.2 expression: 5.1 extent: 3.2 false: 2.4 files: 9.1 formal parameter list: 4 formal: 3.2

frozen value list: 4.4 full item: 7.2 full strips: 8.4 function: 4.4 garbage collection: 7.4 general selector: 7.1 general update routine: 7.1 global declaration: 3.2 global: 3.2 goto statement: 5.4 hole in the extent: 3.2 identifier: 3.1 imperative expression: 5.3 imperative sequence: 5.3 initialisation: 3.2 items: 2.1 label: 5.4 lambda expression: 4.1 link: 8.3 list: 8.3 local declaration: 3.2 local: 3.2 macro: 11.2 meaning: 8.6 *n*-tuple: 4.2 nonlocal: 4.3 null list: 8.3 number: 4.6 opened: 9.1 operands: 5.1 operation: 3.2 operator: 5.1 output local list: 4.1 output local variable: 4.2 pairs: 8.2 partial application: 4.4 primitive data structure class: 7.1 program device: 9.1

record class: 7.2 record: 7.2 references: 8.1 results device: 9.2 results file: 9.2 results: 4.2 routine: 4.1 running: 4.2 scope: 3.2 selector: 7.1 share: 7.1 simple expression: 5.1 simple: 2.1 size: 7.2 source items: 5.5 source: 5.5 specification of a record: stack: 5.3 standard function: 3.2 standard variable: 3.2 standard input device: standard input file: 9.1 standard output device: 9.2 standard output file: 9.2 standardised: 8.6 strip class: 7.3 strip: 7.3 structure constant: 5.1 subscripts: 8.5 text item: 9.1 top of the stack: 5.3 true: 2.4 truthvalues: 2.4 unique name: 3.2 update routine: 7.1 value: 3.1 variable: 3.1 variadic: 4.2 variresult: 4.2 words: 8.6

SYNTAX DEFINITIONS

alphanumeric: 3.1 assignment: 5.5 binary digit: 2.2 binary integer: 2.2 bracket decorator: 1.4 bracket: 1.4 cancellation: 3.3

frozen formals: 4.4

closed expression: 5.1 code instruction: 10.0 compound expression: 5.1 condition: 6.2 conditional body: 6.1 conditional expression: 6.1 constant: 5.1

quasi compound expres-

program file: 9.1

protected: 3.2

sion: 5.5

decimal integer: 2.2 declaration list element: 3.2 declaration: 3.2 decorated bracket: 8.6 destination: 5.5 digit: 1.4 dot operator: 5.1

INDICES

else clause: 6.1 elseif clause: 6.1 exponent: 1.4, 2.3 expression list: 5.1 expression: 5.1 formal parameter list element: 4.1 formal parameter list: 4.1 function body: 4.1 function definition: 4.1 function: 4.1 goto statement: 5.4 identifier: 3.1 imperative sequence: 5.3 imperative: 5.3 integer: 2.2 label: 5.4 labelled statement: 5.4

lambda expression: 4.1

lambda: 4.1

letter: 1.4
list constant brackets: 8.3
list constant element: 8.3
list constant: 8.3
list expression brackets: 8.3
list expression: 8.3
macro definition: 11.2
non-operation identifier: 5.1
octal digit: 2.2
operation: 5.1
output local list: 4.1
output local list element:
4.1

parentheses: 5.1 partial application: 4.4 period: 1.4 program element: 11.1

program: 11.1

destref: 8.1

quasi compound expression:

5.5 auote: 1.4 auoted word: 8.6 real: 2.3 restriction: 3.2 restrictor: 3.2 separator: 1.4 sign: 1.4 simple expression: 5.1 statement: 5.3 string bracket: 8.4 string constant element: 8.4 string constant: 8.4 string quote: 1.4 structure constant: 5.1 structure expression: 5.1 sub ten: 1.4

unquoted word: 8.6

newanyarray: 8.5

STANDARD FUNCTIONS AND VARIABLES

<: 4.6 >: 4.6 =<:4.6 >=:4.6 +:4.6-:4.6*: 4.6 /: 4.6 ↑:4.6 //2.2 =: 2.1, 7.1⇒: 9.2 ::: 8.3 <>: 8.3 arrav: 8.5 atom: 8.2 back: 8.2 booland: 2.4 boolor: 2.4 charin: 9.1 charout: 9.2 charword: 8.6 closefile: 9.1 cons: 8.3 conspair: 8.2 consref: 8.1 consword: 8.6 cont: 8.1 copy: 7.2 cucharout: 9.2 datalist: 7.2 dataword: 7.2 delitem: 7.4

dest: 8.3

destpair: 8.2

destword: 8.6 enddata: 7.2 fnpart: 8.7 fnprops: 8.7 fntolist: 8.3 front: 8.2 frozval: 8.7 hd: 8.3 incharitem: 9.1 init: 8.4 initc: 8,4 intadd: 2.2 intgr: 2.2 intgreq: 2.2 intle: 2.2 intleea: 2.2 intminus: 2.2 intmult: 2.2 intof: 2.3 intplus: 2.2 intsign: 2,2 intsub: 2.2 iscompnd: 2.1 isfunc: 8.7 isinteger: 2.1 isreal: 2.1 itemread: 9.1 logand: 2.2 lognot: 2.2 logor: 2.2 logshift: 2.2 macresults: 11.2 meaning: 8.6

newarray: 8.5 next: 8.3 nextchar: 9.1 nl: 9.2 not: 2.4 null: 8.3 outcharitem: 9.2 outitem: 9.2 partapply: 4.4 popmess: 9.1 popval: 11.3 print: 9.2 proglist: 9.1 realadd: 2.3 realdiv: 2.3 realgr: 2.3 realgreq: 2.3 realle: 2.3 realleeq: 2.3 realmult: 2.3 realminus: 2.3 realof: 2.3 realplus: 2.3 realsign: 2.3 realsub: 2.3 recordfns: 7.2 setpop: 11.3 sp: 9.2 stripfns: 7.3 subscr: 8.4 subscrc: 8.4 tl: 8.3 updater: 8.7

INDICES

SYNTAX WORDS

 :: 5.4
 elseif: 6.1

 →: 5.5
 end: 4.1

 ⇒: 4.1
 function: 3.2

 and: 6.2
 goto: 5.4

 cancel: 3.3
 if: 6.4

 close: 6.1
 lambda: 4.1

 else: 6.1
 macro: 11.2

nonop: 5.1 operation: 3.2 or: 6.2 return: 5.4 routine: 4.1 then: 6.1 vars: 3.2

POP-2 Papers: Introduction and Reference Manual R. M. Burstall, J. S. Collins and R. J. Popplestone

The Department of Machine Intelligence and Perception of the University of Edinburgh has developed POP-2 a programming language for a wide range of computer applications, with special emphasis on non-numerical work.

POP-2 has been designed for on-line use, varying from simple 'scratch-pad' calculations to the solution of problems where complex data structures must be processed. It has been used as a tool for the Department's research work in machine intelligence on topics such as heuristic problem solving, theorem proving and conversational question answering. Advanced features include immediate execution of expressions, a record processing scheme which allows list processing as a special case, 'dynamic' lists and a full treatment of functions. The language, following the CPL and ISWIM languages, has powerful facilities for creating and operating on functions. Consistent use of this basic concept has enabled the language to be simple and easy to learn.

The 'Introduction to POP-2' explains the elementary features of the language assuming little or no knowledge of programming. For the more experienced reader it will also serve as a useful preface for the more formal exposition of the 'Reference Manual'. The latter contains a complete definition of the language with full indices of technical terms, syntax definitions and standard functions.

In an editorial article on artificial intelligence *Nature* said, '... the field on which the group at Edinburgh is embarked has the most exciting promise. Enough has already been done... to demonstrate the great interest of attempts to use computers for tasks other than wooden and repetitive arithmetic... there is the real prospect that everybody will be stimulated and improved by work of the kind now being undertaken.'

The authors are members of the Department of Machine Intelligence and Perception, University of Edinburgh.

Published by Oliver & Boyd Ltd. Tweeddale Court 14 High Street Edinburgh 1